



SDL::Manual

Writing Games in Perl

Kartik Thakore

With contributions by the community

Latex based on the Perl6 book: [HTTPS://GitHub.Com/perl6/book](https://github.com/perl6/book)

Contents

1 Preface	1
1.1 SDL and SDLx	2
1.2 About the Book	4
1.3 Installing SDL Perl	4
1.3.1 Windows	4
1.3.2 Mac OS X	5
1.3.3 GNU/Linux	5
1.3.4 CPAN install	6
1.4 Contact	6
1.5 Examples	7
1.6 Acknowledgements	7
2 The Screen	9
2.1 SDLx::App Options	10
2.1.1 Shortcuts	10
3 Drawing	13
3.1 Coordinates	13
3.2 Drawing with SDL	14

3.2.1	Surface Drawing Methods	15
3.2.2	Pixels	17
3.2.3	Primitives	17
3.2.4	Drawing with Primitives	21
3.3	Drawing on Multiple Surfaces	22
3.3.1	Creating Surfaces	23
3.4	Lots of Flowers but One Seed	23
4	Handling Events	27
4.1	Quitting with Grace	30
4.1.1	Exit on Quit	31
4.2	Small Paint: Input Devices	32
4.2.1	Saving the image	32
4.2.2	Keyboard	33
4.2.3	Mouse	34
4.3	POD ERRORS	35
5	The Game Loop	37
5.1	A Practical Game Loop	38
5.1.1	Fixed FPS	40
5.1.2	Variable FPS	44
5.2	Integrating Physics	44
5.2.1	Laser in Real Time	45
6	Pong!	49
6.1	The Basic Screen	50
6.2	Game Objects	51
6.2.1	Show it Off	52
6.3	Moving the Player's Paddle	53
6.3.1	Rinse and Repeat	54
6.3.2	Move that Paddle!	55

6.4	A Bouncing Ball	57
6.4.1	He shoots... and scores!!	58
6.5	Collision Detection: The Ball and The Paddle	60
6.6	Artificial Stupidity	63
6.7	Cosmetics: Displaying the Score	64
6.8	Exercises	66
6.8.1	Answers	67
6.9	Author	69
6.10	POD ERRORS	69
7	Tetris	71
7.1	The Game Window	72
7.2	Managing Blocks	73
7.3	Piece Collisions	75
7.3.1	Score and Game State	79
7.3.2	Drawing the Game	81
7.4	Author	82
8	Puzz! A puzzle game	83
8.1	Abstract	83
8.2	The Window	84
8.3	Loading the images	85
8.4	Handling Events	87
8.5	Filling the Grid	87
8.6	Moving the Pieces	88
8.6.1	The Move Handler Callback	89
8.7	Rendering the Game	89
8.8	Complete Code	92
8.9	Activities	95
8.10	Author	95

9 Sound and Music	97
9.1 Simple Sound Script	98
9.1.1 Loading Samples	98
9.1.2 Playing the sample and closing audio	99
9.1.3 Streaming Music	100
9.1.4 Code so far	101
9.2 Sound Applications	102
9.2.1 SDLx::App Audio Initialization	102
9.2.2 Loading Resources	103
9.2.3 The Show Handler	104
9.2.4 The Event Handler	105
9.2.5 Completed Code	107
9.3 Music Visualizer	111
9.3.1 The Code and Comments	112
10 CPAN	119
10.1 Modules	119
10.1.1 MVC Method	120
10.2 Picking Modules	121
10.2.1 Documentation	121
10.2.2 License	122
10.2.3 Ratings	123
10.2.4 Dependencies	123
10.2.5 CPAN Testers Charts	124
10.2.6 Release Date	125
10.3 Conclusion	125
10.4 Author	126
11 Pixel Effects	127
11.1 Sol's Ripple Effect	128

11.1.1Pure Perl	128
11.1.2Inline Effects	130
12 Additional Modules	133
12.1PDL	133
12.1.1Make the application	134
12.1.2Attaching the Piddle	135
12.1.3Drawing and Updating	136
12.1.4Running the App	136
12.1.5Complete Program	137
12.2OpenGL and SDL	138
12.2.1SDL Setup	139
12.2.2OpenGL Setup	140
12.2.3The Render Callback	140
12.2.4Event handling	141
12.2.5Complete Code	143
13 Free Resources	145
13.1Art and Sprites	146
13.2Music and Sound Effects	147
13.3Fonts	147
13.4DIY	148
13.5Author	148

1

Preface

Simple DirectMedia Layer (or *libsdl*) is a cross-platform C library that provides access to several input and output devices. Its most popular usage is to provide access to the video framebuffer and input devices for games. SDL also has several extension libraries to provide features such as text display, sound mixing, image handling, and graphics effects.

SDL Perl binds several of these libraries together in the `SDL::*` namespace. Moreover, SDL Perl provides several high-level libraries in the `SDLX::*` namespace that encapsulate valuable game-writing abstractions.

1.1 SDL and SDLx

The main purpose of the `SDLx::*` layer is to smooth out the drudgery of using the `SDL::*` layer directly.

Don't worry about understanding the details of this code right now. Compare the complexity and size of the code listings.

Using the `SDL::*` layer to draw a blue rectangle looks something like:

```
1  use SDL;
2  use SDL::Video;
3  use SDL::Surface;
4  use SDL::Rect;
5
6  # the size of the window box or the screen resolution if fullscreen
7  my $screen_width  = 800;
8  my $screen_height = 600;
9
10 SDL::init(SDL_INIT_VIDEO);
11
12 # setting video mode
13 my $screen_surface = SDL::Video::set_video_mode($screen_width,
14                                                  $screen_height,
15                                                  32,
16                                                  SDL_ANYFORMAT);
17
18 # drawing a rectangle with the blue color
19 my $mapped_color = SDL::Video::map_RGB($screen_surface->format(),
20                                       0, 0, 255);
21 SDL::Video::fill_rect($screen_surface,
22                      SDL::Rect->new($screen_width / 4, $screen_height / 4,
23                                     $screen_width / 2, $screen_height / 2),
24                      $mapped_color);
```

```

25
26 # update an area on the screen so it's visible
27 SDL::Video::update_rect($screen_surface, 0, 0,
28                          $screen_width, $screen_height);
29
30 # just to have time to see it
31 sleep(5);

```

... while drawing a blue rectangle in the `SDLx::*` layer is as simple as:

```

1  use strict;
2  use warnings;
3
4  use SDL;
5  use SDLx::App;
6
7  my $app = SDLx::App->new( width=> 800, height => 600 );
8
9  $app->draw_rect([ $app->width / 4, $app->height / 4,
10                  $app->width / 2, $app->height / 2, ],
11                [ 0, 0, 255, 255] );
12
13  $app->update();
14
15  sleep(5);

```

The `SDLx::*` modules also provide and manage higher-level concerns for users, such as layers and game loops.

1.2 About the Book

This book has a two-fold purpose: first, to introduce game development to Perl programmers, and second, to introduce Modern Perl concepts through game development. While the examples assume some experience with Perl, no experience with SDL in Perl or as `libSDL` itself is necessary.

The book presents a progression from simple to intermediate examples and provides suggestions for more advanced endeavors. The chapters of this book increase progressively in complexity, but each chapter has a singular goal (such as chapter five's *Making Pong*) which stands alone as an individual tutorial. Sources and data files are all available from <http://SDL.perl.org/>.

1.3 Installing SDL Perl

We assume the presence of a recent version of the Perl language (at least Perl 5.10) and supporting packages. We also assume that you can install packages from the CPAN, including SDL Perl itself.

1.3.1 Windows

`Alien::SDL` will install binaries for 32bit and 64bit so there is no need to compile anything.

1.3.2 Mac OS X

Fink has packages for SDL Perl available. However, they do not support Pango, a library which provides internalization support for text handling.

Installing `Alien::SDL` from the CPAN will compile SDL and its dependencies, provided you have installed several necessary dependencies. We recommend that you install `libfreetype6`, `libx11`, `libvorbis`, `libogg`, `libpng`, and their headers.

1.3.3 GNU/Linux

Most current GNU/Linux distributions include all the parts needed for this tutorial in the default install and in their package management system. It is also always possible to install on GNU/Linux using the available open source code from the proper repositories. The `Alien::SDL` perl module automates much of downloading, compiling, and installing the needed libraries.

You can probably use your distribution's packages. On Ubuntu and Debian try:

```
$ sudo apt-get install libsdl-net1.2-dev libsdl-mixer1.2-dev \
libsdl1.2-dev libsdl-image1.2-dev libsdl-ttf2.0-dev \
libsdl-gfx1.2-dev libsdl-pango-dev
```

To compile from scratch, you must install a compiler, system header packages, and some libraries are required.

```
$ sudo apt-get install build-essential xorg-dev libx11-dev libxv-dev \
libpango1.0-dev libfreetype6-dev libvorbis-dev libpng12-dev \
libogg-dev
```

1.3.4 CPAN install

Before installing SDL Perl, ensure that you have the most recent versions of the modules necessary to build SDL:

```
$ sudo cpan CPAN
$ sudo cpan YAML Module::Build
```

After these two steps CPAN will be able to install SDL:

```
$ sudo cpan SDL
```

For most platforms a CPAN install will suffice. Supported and tested platforms are listed at <http://pass.cpantesters.org/distro/S/SDL.html>.

1.4 Contact

Hopefully this book answers most of your questions. For additional assistance, contact the project via:

- *the web*, by visiting the SDL Perl homepage at <http://sd1.perl.org/>.
- *IRC*, in the #sd1 channel on irc.perl.org. This is a very active and helpful resource.
- *email*, through the sd1-devel@perl.org mailing list.

1.5 Examples

The code examples in this book are available from https://github.com/PerlGameDev/SDL_Manual/tree/master/code_listings.

1.6 Acknowledgements

Thanks to contributors and reviewers from the #sdl channel, including:

Alias

bobross

Blaizer

cfedde

chromatic

FROGGS

garu

jamesw

perlpilot

PerlJam

Pip

waxhead

Chapter 1 | PREFACE

and many more

(Apologies if I have missed you; let me know and I will add you.)

2

The Screen

SDL's primary purpose is to display graphics. It does so by providing an abstraction called a *screen*, which represents a *video device*. This video device is an interface provided by your operating system, such as X11 or DirectX. Before you can display anything, you must create a screen. The `SDLx::App` class does so for you:

```
1  use strict;
2  use warnings;
3  use SDL;
4  use SDLx::App;
5
6  my $app = SDLx::App->new();
7
8  sleep( 2 );
```

This example causes an empty window to appear on the desktop. Most systems will fill that window with the color black. Other systems might display a transparent window. SDL's

Chapter 2 | THE SCREEN

default behavior is to fill the screen with black. To enforce this behavior on all systems, you must `update()` the app to draw to the window:

```
$app->update();
```

2.1 SDLx::App Options

`SDLx::App` allows you to specify several options for the screen and your application. First are the physical dimensions of the screen itself. To make the screen of the `SDLx::App` window a 400×400 pixel square, change the initialization line to:

```
my $app = SDLx::App->new( width => 400, height => 400 );
```

Another important option is the window's title. Some systems display the path to the running program. Others leave the title blank. You can change the displayed title with another argument to the `SDLx::App` constructor:

```
my $app = SDLx::App->new( width  => 400,  
                          height => 400,  
                          title  => 'Pong - A clone' );
```

At this point your screen will be:

2.1.1 Shortcuts

Abbreviations for these parameters are available. Instead of `width`, `height`, and `title`, you may use `w`, `h`, and `t` respectively. The previous example could also be written:

```
my $app = SDLx::App->new( w => 400,  
                          h => 400,  
                          t => 'Pong - A clone' );
```




Figure 2.1: Your first SDL screen!

3

Drawing

SDL provides several ways to draw graphical elements on the screen in three general categories: primitives, images, and text. All drawing occurs on a surface, represented by the `SDLx::Surface` class. Even the `SDLx::App` is an `SDLx::Surface`. Though this means it's possible to draw directly to the app's surface, there are several advantages to drawing on multiple surfaces.

3.1 Coordinates

SDL's surface coordinate system has its origin (where both the x and y coordinates have the value of zero) in the upper left corner. As the value of x increases, the position moves to the right of the origin. As the value of y increases, the position moves downward from the origin. The API always lists coordinates in x, y order.

The SDL library documentation has an extended discussion on coordinates: <http://sdl-tutorials.com/sdl-coordinates-and-blitting>.

3.2 Drawing with SDL

You can produce original pictures knowing little more than how to draw to a surface with SDL:



Figure 3.1: A field of flowers

3.2.1 Surface Drawing Methods

As mentioned earlier, all drawing in SDL requires a surface. The `SDLx::Surface` object provides access to methods in the form of:

```
$surface->draw_{something}( .... );
```

Parameters to these methods are generally coordinates and colors, provided as array references.

Rectangular Parameters

Some parameters are sets of coordinate positions and dimensions. For example, parameters to describe a rectangle of 40x40 pixels placed at (20, 20) pixel units on the screen make a four-element array reference of x, y, width, height:

```
my $rect = [20, 20, 40, 40];
```

Color

SDL color parameters require four-element array references. The first three numbers define the Red, Green, and Blue intensity of the color. The final number defines the transparency of the color.

```
my $color = [255, 255, 255, 255];
```

The magnitude of each color value determines how much of that color component will be mixed into the resulting color. A 0 value specifies that none of the color channel should be used while 255 specifies a maximum intensity for a particular channel. The first value corresponds with the Red channel, so a higher number there means more red will be mixed into the resulting color. It is a common practice to achieve a grayscale of varying intensity by

Chapter 3 | DRAWING

specifying the same value for each of the Red, Green, and Blue color channels. The fourth and final value designates the transparency (or Alpha channel) where a 0 value makes the resulting color fully transparent and 255 makes it entirely opaque. A transparency value somewhere in between will allow underlying (pixel data of surfaces below the current one) colors to be blended with the specified RGB values into the final color output.

You may also represent a color as hexadecimal values, where the values of the numbers range from 0-255 for 32 bit depth in RGBA format:

```
my $color = 0xFFFFFFFF;  
my $white = 0xFFFFFFFF;  
my $black = 0x000000FF;  
my $red   = 0xFF0000FF;  
my $green = 0x00FF00FF;  
my $blue  = 0x0000FFFF;
```

... or as four-byte hexadecimal values, where each two-digit byte encodes the same RGBA values:

```
my $goldenrod = 0xDAA520FF;
```

NOTE: Depth of Surface

The color depth of the surface—how many bits are available to describe colors—is a property of the relevant `SDLx::Surface` or `SDLx::App`. Set it in its constructor:

```
my $app = SDLx::App->new( depth => 32 );
```

The default bit depth is 32, such that each color component has 256 possible values. Other options are 24, 16, and 8.

3.2.2 Pixels

All `SDLx::Surfaces` are collections of pixels. You can read from and write to these pixels by treating the surface as an array reference:

```
$app->[$x][$y] = $color;
```

... where `$color` is an unsigned integer value using the hexadecimal format (`0xRRGGBBAA`) or an anonymous array of the form `[$red, $green, $blue, $alpha]`.

3.2.3 Primitives

Drawing primitives are simple shapes that SDL supports natively.




Figure 3.2: Drawing a line

Chapter 3 | DRAWING

Lines

A line is a series of contiguous pixels between two points. The `draw_line` method causes SDL to draw a line to a surface:

```
$app->draw_line( [200, 20], [20, 200], [255, 255, 0, 255] );
```

This will draw a yellow line from positions (200, 20) to (20, 200).




Figure 3.3: Drawing a Rectangle

Rectangles

A rectangle is a four-sided, filled polygon. Rectangles are a common building block for games. In SDL, rectangles are the most cost effective of the primitives to draw. The `draw_rect` method draws a rectangle on a surface:

```
$app->draw_rect( [10, 20, 40, 40 ], [255, 255, 255,255] );
```

This draws a white square of size 40x40 onto the screen at the position (10,20).




Figure 3.4: Drawing a Circle




Figure 3.5: Drawing a filled Circle

Circles

A circle is a primitive a fixed radius around a given point. Circles may be filled or unfilled. The `draw.circle` and `draw.circle.filled` methods draw these to a surface:

```
$app->draw_circle(      [100, 100], 20, [255, 0,  0, 255] );  
$app->draw_circle_filled( [100, 100], 19, [0,  0, 255, 255] );
```

These draw an unfilled red circle and a filled blue circle.

SDL provides more complex primitives in `SDL::GFX::Primitives`.

3.2.4 Drawing with Primitives

It's easy to combine several primitives to draw an interesting images.

```
1  use strict;
2  use warnings;
3  use SDL;
4  use SDLx::App;
5
6  my $app = SDLx::App->new(
7      w      => 500,
8      h      => 500,
9      d      => 32,
10     title => 'Pretty Flowers'
11 );
12
13 # Add the blue skies
14 $app->draw_rect( [ 0, 0, 500, 500 ], [ 20, 50, 170, 255 ] );
15
16 # Draw a green field
17 $app->draw_rect( [ 0, 400, 500, 500 ], [ 50, 170, 20, 100 ] );
18
19 # Make a surface for the flower
20 my $flower = SDLx::Surface->new( width => 50, height => 100 );
21
22 # With a black background
23 $flower->draw_rect( [ 0, 0, 50, 100 ], [ 0, 0, 0, 0 ] );
24
25 # Draw a pretty green stem
26 $flower->draw_rect( [ 23, 30, 4, 100 ], [ 0, 255, 0, 255 ] );
27
28 # And a simple flower bud
29 $flower->draw_circle_filled( [ 25, 25 ], 10, [ 150, 0, 0, 255 ] );
30 $flower->draw_circle( [ 25, 25 ], 10, [ 255, 0, 0, 255 ] );
31
32 # Draw flower on $app
33 $flower->blit( $app, [ 0, 0, 50, 100 ] );
```

Chapter 3 | DRAWING

```
34  
35  $app->update();  
36  
37  sleep(1);
```




Figure 3.6: Looks so lonely there all alone

3.3 Drawing on Multiple Surfaces

The examples so far have drawn on only a single surface, the display. SDL makes it possible to write on multiple surfaces. These other surfaces exist only in memory until you draw them to the display.

3.3.1 Creating Surfaces

There are several ways to create an `SDLx::Surface` for use. The most common is to create one manually with a constructor call:

```
$surface = SDLx::Surface->new( width => $width, height => $height );
```

`SDL::Image` and `SDL::Video` can load images as surfaces too. `SDL::Image` provides support for all types of images, provided that the underlying `SDL_image` library supports the image type you want to load. For example, `SDL_image` must support PNG images to use:

```
$surface = SDL::Image::load( 'picture.png' );
```

In the event that the desired `SDL_image` library is unavailable, you can fallback to the built-in support for the `.bmp` format.

```
$surface = SDL::Video::load_BMP( 'picture.bmp' );
```

The `SDLx::Sprite` module provides another option to manipulate surfaces.

3.4 Lots of Flowers but One Seed

The flower example used a method called `blit` to draw a surface to the display. This method copies data from one surface to another. It's a fundamental operation, but it's a low level operation. `SDLx::Sprite` provides higher level options. Besides making drawing simpler, `SDLx::Sprite` adds several other features useful for moving images. Here's a revised example using `SDLx::Sprite` for flowers:

```
1 use strict;
2 use warnings;
3 use SDL;
4 use SDLx::App;
```

Chapter 3 | DRAWING

```
5 use SDLx::Sprite;
6
7 my $app = SDLx::App->new(
8     w      => 500,
9     h      => 500,
10    d       => 32,
11    title => 'Pretty Flowers'
12 );
13
14 # Adding blue skies
15 $app->draw_rect( [ 0, 0, 500, 500 ], [ 20, 50, 170, 255 ] );
16
17 # Draw a green field
18 $app->draw_rect( [ 0, 400, 500, 100 ], [ 50, 170, 20, 100 ] );
19
20 my $flower = SDLx::Sprite->new( width => 50, height => 100 );
21
22 # Use ->surface() to access a sprite's SDLx::Surface
23
24 # Make the background black
25 $flower->surface->draw_rect( [ 0, 0, 50, 100 ], [ 0, 0, 0, 0 ] );
26
27 # Now for a pretty green stem
28 $flower->surface->draw_rect( [ 23, 30, 4, 100 ], [ 0, 255, 0, 255 ] );
29
30 # Add the simple flower bud
31 $flower->surface->draw_circle_filled( [ 25, 25 ], 10, [ 150, 0, 0, 255 ] );
32 $flower->surface->draw_circle( [ 25, 25 ], 10, [ 255, 0, 0, 255 ] );
33
34 $flower->draw_xy( $app, 0, 0 );
35
36 $app->update();
37
38 sleep(1);
```

Flowers usually don't grow in the sky. Flowers make more sense on the ground. It's easy to insert plenty of identical flowers from a single sprite. Replace the line:


```
$flower->draw_xy( $app, 0, 0 );
```

... with:

```
1  for (0 .. 500) {  
2      my $y = 425 - rand( 50);  
3      $flower->draw_xy( $app, rand(500) - 20, $y );  
4  }
```

... to make an entire field of flowers.

4

Handling Events

The cornerstone of an SDL application is event handling. The user presses a key or moves the mouse. The operating system switches the focus of the active window. The user selects the quit option from the menu or the operating system. These are all events. How do you handle them?

SDL provides an event queue which holds all events that occur until they are removed. Every time an event occurs, SDL places it into the queue. The `SDL::Event` object represents this queue in Perl, allowing you to add and remove events constantly:

```
1      use strict;
2      use warnings;
3      use SDL;
4      use SDL::Event;
5      use SDL::Events;
6      use SDLx::App;
7
```

Chapter 4 | HANDLING EVENTS

```
8      my $app    = SDLx::App->new( w => 200, h => 200 );
9      my $event = SDL::Event->new();
10
11      my $quit   = 0;
12
13      while (!$quit) {
14          # Updates the queue to recent events
15          SDL::Events::pump_events();
16
17          # process all available events
18          while ( SDL::Events::poll_event($event) ) {
19
20              # check by Event type
21              do_key() if $event->type == SDL_KEYDOWN;
22          }
23      }
24
25      sub do_key { $quit = 1 }
```

Every event has an associated type which represents the category of the event. The previous example looks for a keypress event ¹. The SDL library defines several types of events, and `SDL_perl` makes them available as constants with names such as `SDL_KEYDOWN` and `SDL_QUIT`. See `perldoc SDL::Events` for a list of all event types.

Checking for every possible event type within that event loop can be tedious. The `SDLx::Controller` available from the `SDLx::App` offers the use of event callbacks with which to handle events. Processing events is a matter of setting up the appropriate callbacks and letting SDL do the heavy work.

SDL Events Types

Additional Event types that can be captured by SDL are:

Keyboard

¹ SDL separates the event of pressing a key from the event of releasing a key, which allows you to identify combinations of keypresses, such as Ctrl + P to print.

SDL.KEYDOWN SDL.KEYUP - Keyboard button pressed

Mouse

SDL.MOUSEMOTION - Mouse motion occurred

SDL.MOUSEBUTTONDOWN SDL.MOUSEBUTTONUP - Mouse button pressed

Joystick

SDL.JOYAXISMOTION - Joystick axis motion

SDL.JOYBALLMOTION - Joystick trackball motion

SDL.JOYHATMOTION - Joystick hat position change

SDL.JOYBUTTONDOWN SDL.JOYBUTTONUP - Joystick button pressed

Window & System

SDL.ACTIVEEVENT - Application visibility

SDL.VIDEORESIZE - Window resized

SDL.VIDEOEXPOSE - Window exposed

SDL.QUIT - Quit requested

SDL.USEREVENT - A user-defined event type

SDL.SYSWMEVENT - Platform-dependent window manager event

For more information look at:

```
perldoc SDL::Event
```

4.1 Quitting with Grace

The example applications so far have not exited cleanly. Handling quit events is much better:

```
1  use strict;
2  use warnings;
3  use SDL;
4  use SDL::Event;
5  use SDLx::App;
6
7  my $app = SDLx::App->new(
8      w      => 200,
9      h      => 200,
10     d      => 32,
11     title => "Quit Events"
12 );
13
14 $app->add_event_handler( \&quit_event );
15 $app->run();
16
17 sub quit_event
18 {
19     # the callback receives the appropriate SDL::Event
20     my $event = shift;
21
22     # ... as well as the calling SDLx::Controller
23     my $controller = shift;
24
25     # stopping the controller will exit $app->run() for us
```

```
26         $controller->stop if $event->type == SDL_QUIT;
27     }
```

`SDLx::App` calls the `event_handlers`, from an internal `SDLx::Controller`. When this event handler receives a quit event, it calls `SDLx::Controller::stop()` which causes `SDLx::App` to exit gracefully.

4.1.1 Exit on Quit

Exiting on receiving the `SDL_QUIT` event is such a common operation that `SDLx::App` provides it as a constructor option:

```
1  use strict;
2  use warnings;
3  use SDL;
4  use SDLx::App;
5
6  my $app = SDLx::App->new(
7      w      => 200,
8      h      => 200,
9      d      => 32,
10     title   => "Quit Events",
11     exit_on_quit => 1
12 );
13
14 $app->run();
```

4.2 Small Paint: Input Devices

SDL events also allow input handling. Consider a simple paint program. It will provide a small black window. Moving the mouse draws on this window. Pressing a number key chooses a paint color. Pressing q or Q exits the program. Pressing c or C clears the screen. Pressing ctrl-S saves the image to a file named *painted.bmp*.



Figure 4.1: Simple Paint: Smile

4.2.1 Saving the image

Start by defining the saving function:

```
1 sub save_image {  
2     if (SDL::Video::save_BMP( $app, 'painted.bmp' ) == 0  
3         && -e 'painted.bmp')
```



```

4      {
5          warn 'Saved painted.bmp to ' . cwd();
6      }
7      else
8      {
9          warn 'Could not save painted.bmp: ' . SDL::get_errors();
10     }
11 }

```

4.2.2 Keyboard

Keyboard handling requires some color data as well as a keypress callback:

```

1  my $brush_color = 0;
2
3  sub keyboard_event
4  {
5      my $event = shift;
6
7      if ( $event->type == SDL_KEYDOWN )
8      {
9          # convert the key_symbol (integer) to a keyname
10         my $key_name = SDL::Events::get_key_name( $event->key_sym );
11
12         # if $key_name is a digit, use it as a color
13         $brush_color = $key_name if $key_name =~ /\d$/;
14
15         # get the keyboard modifier (see perldoc SDL::Events)
16         my $mod_state = SDL::Events::get_mod_state();
17
18         # we are using any CTRL so KMOD_CTRL is fine
19         save_image() if $key_name =~ /\s$/ && ($mod_state & KMOD_CTRL);
20
21         # clear the screen
22         $app->draw_rect( [ 0, 0, $app->w, $app->h ], 0 )
23         if $key_name =~ /\c$/;

```

Chapter 4 | HANDLING EVENTS

```
24
25     # exit
26     $app->stop() if $key_name =~ /^q$/;
27 }
28
29 $app->update();
30 }
31
32 $app->add_event_handler(\&quit_event);
33 $app->add_event_handler(\&keyboard_event);
```

NOTE: When adding a callback to `SDLx::App` which uses variables declared outside of the function (`$brush_color` and `@colors` in this case), be sure to define them before declaring the subroutine. Normal Perl scoping and initialization rules apply.

4.2.3 Mouse

Handling mouse events is almost as straightforward as keyboard events: =begin program-listing

```
# track the drawing status
my $drawing = 0;

sub mouse_event {
    my $event = shift;

    # detect Mouse Button events and check if user is currently drawing
    if ($event->type == SDL_MOUSEBUTTONDOWN || $drawing)
    {
        # set drawing to 1
        $drawing = 1;

        # get the X and Y values of the mouse
```

```

        my $x = $event->button_x;
        my $y = $event->button_y;

        # draw a rectangle at the specified position
        $app->draw_rect( [ $x, $y, 2, 2 ], $colors[$brush_color] );

        $app->update();
    }

    # disable drawing when user releases mouse button
    $drawing = 0 if ( $event->type == SDL_MOUSEBUTTONDOWN );
}

$app->add_event_handler( \&mouse_event );

```

This is all of the code necessary to make a simple drawing application.

Take note of two things. First, `SDL_perl` invokes the event handlers in the order of attachment. If the user presses `q` and then moves the mouse, the application will quit before processing the mouse movement.

Second, the application makes no distinction between right, middle, or left mouse clicks. SDL provides this information. See the `button_button()` method in `SDL::Event`.

4.3 POD ERRORS

Hey! The above document had some coding errors, which are explained below:

Around line 317:

```
=end programlisting without matching =begin. (Stack: [empty])
```


5

The Game Loop

Just as an interactive SDL app builds around an event loop, a game builds around a game loop. The simplest game loop is something like:

```
1 while (!$quit)
2 {
3     get_events();
4     calculate_next_positions();
5     render();
6 }
```

The names of the functions called in this loop hint at their purposes, but the subtleties of even this simple code are important. `get_events()` obviously processes events from the relevant input devices (keyboard, mouse, joystick). Processing events at the start of every game loop iteration helps to prevent lag.

`calculate_next_positions` updates the game state according to user input as well as any active animations (a player walking, an explosion, a cut scene). `render()` finally updates and displays the screen.

5.1 A Practical Game Loop

Consider a game with a moving laser bolt:

```
1  use strict;
2  use warnings;
3  use SDL;
4  use SDL::Event;
5  use SDL::Events;
6  use SDLx::App;
7
8  my $app = SDLx::App->new(
9      width => 200,
10     height => 200,
11     title => 'Pew Pew'
12 );
13
14 my $quit = 0;
15
16 # start laser on the left
17 my $laser = 0;
18
19 sub get_events {
20     my $event = SDL::Event->new();
21
22     SDL::Events::pump_events;
23
24     while( SDL::Events::poll_event($event) )
25     {
26         $quit = 1 if $event->type == SDL_QUIT
27     }
```

```

28 }
29
30 sub calculate_next_positions {
31     # move the laser
32     $laser++;
33
34     # if the laser goes off the screen, bring it back
35     $laser = 0 if $laser > $app->w();
36 }
37
38 sub render {
39     # draw the background first
40     $app->draw_rect( [ 0, 0, $app->w, $app->h ], 0 );
41
42     # draw the laser halfway up the screen
43     $app->draw_rect( [ $laser, $app->h / 2, 10, 2 ], [ 255, 0, 0, 255 ]);
44
45     $app->update();
46 }
47
48 while (!$quit)
49 {
50     get_events();
51     calculate_next_positions();
52     render();
53 }

```

This game loop works very well for consoles and other devices where you know exactly how much CPU time the game will get for every loop iteration. That hardware stability is easy to predict: each animation and calculation will happen at the same time for each machine. Unfortunately, this is *not* true for modern operating systems and general purpose computing hardware. CPU speeds and workloads vary, so for this game to play consistently across multiple machines and myriad configurations, the game loop itself needs to regulate its updates.

5.1.1 Fixed FPS

One way to solve this problem is to regulate the number of frames per second the game will produce. A *frame* is a complete redraw of the screen representing the updated game state. If each iteration of the game loop draws one frame, the more frames per second, the faster the game is running. If the game loop limits the number of frames per second, the game will perform consistently on all machines fast enough to draw that many frames per second.

You can see this with the example program *game_fixed.pl*. When run with no arguments:

```
$ perl game_fixed.pl
```

.... the FPS rate will be erratic. The laser seems to change its speed randomly. When run with a single argument, the game sets an upper bound on the number of frames per second:

```
$ perl game_fixed.pl 1
```

This will prevent the laser from going faster than 60 frames per second. When run with a second argument, the game will set a lower bound of frames per second:

```
$ perl game_fixed.pl 1 1
```

At this point the FPS should hold steady at 60 frames per second.

```
1      use strict;
2      use warnings;
3      use SDL;
4      use SDL::Event;
5      use SDL::Events;
6      use SDLx::App;
7
8      my $app = SDLx::App->new(
9          width => 200,
```



```

10             height => 200,
11             title  => 'Pew Pew'
12         );
13
14     my ( $start, $end, $delta_time, $FPS, $frames ) = ( 0, 0, 0, 0, 0 );
15
16     # aim for a rate of 60 frames per second
17     my $fixed_rate = 60;
18
19     # compensate for times stored in microseconds
20     my $fps_check = (1000 / $fixed_rate );
21
22     my $quit = 0;
23
24     # start laser on the left
25     my $laser = 0;
26
27     sub get_events {
28         my $event = SDL::Event->new();
29
30         SDL::Events::pump_events;
31
32         while ( SDL::Events::poll_event($event) ) {
33             $quit = 1 if $event->type == SDL_QUIT;
34         }
35     }
36
37     sub calculate_next_positions {
38         $laser++;
39
40         $laser = 0 if $laser > $app->w;
41     }
42
43     sub render {
44         # draw the background first
45         $app->draw_rect( [ 0, 0, $app->w, $app->h ], 0 );
46
47         # draw the laser

```

Chapter 5 | THE GAME LOOP

```
48         $app->draw_rect( [ $laser, $app->h / 2, 10, 2 ], [ 255, 0, 0, 255 ] );
49
50         # draw the FPS
51         $app->draw_gfx_text( [ 10, 10 ], [ 255, 0, 255, 255 ], "FPS: $FPS" );
52
53         $app->update();
54     }
55
56     # Called at the end of each frame, whether we draw or not
57     sub calculate_fps_at_frame_end
58     {
59         # Ticks are microseconds since load time
60         $end = SDL::get_ticks();
61
62         # smooth the frame rate by averaging over 10 frames
63         if ( $frames < 10 ) {
64             $frames++;
65             $delta_time += $end - $start;
66         }
67         else {
68             # frame rate is Frames * 100 / Time Elapsed in us
69             $FPS = int( ( $frames * 100 ) / $delta_time )
70             if $delta_time != 0;
71
72             # reset metrics
73             $frames = 0;
74             $delta_time = 0;
75         }
76     }
77
78     while ( !$quit ) {
79         # Get the time for the starting of the frame
80         $start = SDL::get_ticks();
81
82         get_events();
83
84         # if fixing the lower bounds of the frame rate
85         if( $ARGV[1] )
```

```

86         {
87             # if delta time is going too slow for frame check
88             if ( $delta_time > $fps_check ) {
89
90                 calculate_fps_at_frame_end();
91
92             # skip rendering and collision detections
93             # (heavy functions in the game loop)
94                 next;
95             }
96         }
97
98         calculate_next_positions();
99         render();
100
101         # a normal frame with rendering actually performed
102         calculate_fps_at_frame_end();
103
104         # if fixing the upper bounds of the frame rate
105         if ( $ARGV[0] ) {
106
107             # if delta time is going too fast compared to the frame check
108             if ( $delta_time < $fps_check ) {
109
110                 # delay for the difference
111                 SDL::delay( $fps_check - $delta_time );
112             }
113         }
114     }

```

This method is generally sufficient for most computers. The animations will be smooth enough to provide the same gameplay even on machines with different hardware.

However, this method still has some serious problems. First, if a computer is too slow to sustain a rate of 60 FPS, the game will skip rendering some frames, leading to sparse and jittery animation. It will skip a lot of rendering, and the animation will look sparse and

jittery. It might be better to set a lower bounds of 30 FPS, though it's difficult to predict the best frame rate for a user.

The worst problem is that this technique still ties rendering speed to the CPU speed: a very fast computer will waste CPU cycles delaying.

5.1.2 Variable FPS

To fix the problem of a computer being consistently too fast or too slow for the hard-coded FPS rate is to adjust the FPS rate accordingly. A slow CPU may limit itself to 30 FPS, while a fast CPU might run at 300 FPS. Although you may achieve a consistent rate this way (consistent for any one particular computer), this technique still presents the problem of differing animation speeds between different computers.

Better solutions are available.

5.2 Integrating Physics

The problem caused by coupling rendering to the CPU speed has a convenient solution. Instead of updating object positions based on how fast the computer can get through the game loop, derive their positions from a physical model based on the passage of time. Objects moving according to real world time will have consistent behavior at all CPU speeds and smooth interpolation between frames. `SDLx : : App` provides this behavior through movement and show handlers.

Consider a simple physics model for the laser has a consistent horizontal velocity in pixels per time step at the window's mid-point:

```
X = Velocity * time step,  
Y = 100
```

Assuming a velocity of 10, the laser will pass through the coordinates:

```
0, 100
10, 100
20, 100
30, 100
...
200, 100
```

Note that the speed of processing the game loop no longer matters. The position of the laser depends instead on the passage of real time.

The biggest problem with this approach is the required bookkeeping for the many objects and callbacks. The implementation of such complex models is non-trivial; see the lengthy discussion in the documentation of the `SDLx::Controller` module.

`SDLx::App` using the `SDLx::Controller` module provide callbacks to handle both aspects of this type of game loop. One is the movement handler, which is a callback where calculations of the next step for each relevant data point is calculated. In the above example the movement handler would calculate the `x` and `y` values, for each time step between the frames of animations.

When we are ready to render the frame it is handled by the show handler. In the above example that would mean the show handler would print or render the `x`, `y` values.

5.2.1 Laser in Real Time

This version of the laser example demonstrates the use of movement, show handlers, and a simple physics model. This example also shows how `SDLx::App` can do more of the work, even providing the entire game loop:

```
1    use strict;
2    use warnings;
3    use SDL;
```

Chapter 5 | THE GAME LOOP

```
4      use SDL::Event;
5      use SDLX::App;
6
7      my $app = SDLX::App->new(
8          width => 200,
9          height => 200,
10         title => 'Pew Pew'
11     );
12
13     my $laser = 0;
14     my $velocity = 10;
15
16     $app->add_event_handler( \&quit_event );
17
18     # tell app to handle the appropriate times to
19     # call both rendering and physics calculation
20
21     $app->add_move_handler( \&calculate_laser );
22     $app->add_show_handler( \&render_laser );
23
24     $app->run();
25
26     sub quit_event {
27         my $event = shift;
28         my $controller = shift;
29
30         $controller->stop if $event->type == SDL_QUIT;
31     }
32
33     sub calculate_laser {
34
35         # The step is the difference in Time calculated for the next jump
36         my ( $step, $app, $t ) = @_;
37         $laser += $velocity * $step;
38         $laser = 0 if $laser > $app->w;
39     }
40
41     sub render_laser {
```

```
42     my ( $delta, $app ) = @_;  
43  
44     # The delta can be used to render blurred frames  
45  
46     # draw the background first  
47     $app->draw_rect( [ 0, 0, $app->w, $app->h ], 0 );  
48  
49     # draw the laser  
50     $app->draw_rect( [ $laser, $app->h / 2, 10, 2 ], [ 255, 0, 0, 255 ] );  
51     $app->update();  
52 }
```

To learn more about this topic please, see an excellent blog post by **GafferOnGames.com**:
[HTTP://GafferOnGames.Com/game-physics/fix-your-timestep](http://GafferOnGames.Com/game-physics/fix-your-timestep).

6

Pong!

Pong is one of the first popular video games in the world. Allan Alcorn created it for Atari, Inc. Its release in 1972 was both Atari's first game ever and the spark which began the video game industry.

Pong simulates a table tennis match ("ping pong"). Each player controls a paddle which moves vertically on the screen to hit a ball bouncing back and forth between the players. You earn a point if your opponent is unable to return the ball to your side of the screen.

You can recreate Pong yourself with Perl and SDL.

6.1 The Basic Screen

Start by making a simple screen for Pong. Open a file in your favourite text editor and type:

```
1  #!/usr/bin/perl
2  use strict;
3  use warnings;
4
5  use SDL;
6  use SDLx::App;
7
8  # create the main screen
9  my $app = SDLx::App->new(
10     width      => 500,
11     height     => 500,
12     title      => 'My Pong Clone!',
13     dt         => 0.02,
14     exit_on_quit => 1,
15 );
16
17 # let's roll!
18 $app->run;
```

Save this file as *pong.pl* and run it by typing on the command line:

```
perl pong.pl
```

You should see a 500x500 black window entitled “*My Pong Clone!*”. The only new feature you might not have seen before is the `dt` parameter to the `SDLx::App` constructor. This represents the length, in seconds, of a movement step as managed by an `SDLx::Controller` object. Because the `SDLx::App` object is also an `SDLx::Controller` object, it can handle `SDL_QUIT` events. See chapter 4 for more information.

6.2 Game Objects

There are three main game objects in Pong: two player paddles and the bouncing ball. Paddles are rectangles moving which move vertically. They're easy to represent with `SDLx::Rect` objects. First, put `SDLx::Rect` in your module's declarations:

Next, add a hash reference to store the first player's paddle. Using a hash reference allows the possibility of adding more information later. In a more complex game, consider using an actual object which *contains* an `SDLx::Rect`. For now, this will suffice:

```
1 my $player1 = {
2     paddle => SDLx::Rect->new( 10, $app->h / 2, 10, 40 ),
3 };
4
5 # let's roll!
6 $app->run;
```

This creates a 10x40 paddle rect for the first player on the left side of the screen ($x = 10$) and somewhat in the center ($y = \$app->h / 2$). The second player's paddle is similar:

```
my $player2 = {
    paddle => SDLx::Rect->new( $app->w - 20, $app->h / 2, 10, 40 ),
};

# let's roll!
$app->run;
```

The second paddle needs to appear on the right side of the screen, so its x position is the screen's width minus 20. As the paddle has a width of 10 and the x position refers to the rect's top-left corner, the paddle has a 10 pixel margin from the right edge of the screen.

Finally, the bouncing ball is a 10x10 rect in the middle of the screen:

Chapter 6 | PONG!

```
my $ball = {  
    rect => SDLx::Rect->new( $app->w / 2, $app->h / 2, 10, 10 ),  
};    # let's roll!    $app->run;
```

Just like the original Pong, this ball is square.

6.2.1 Show it Off

With the game objects created, add a show handler to render them to the screen:

```
1  $app->add_show_handler(>  
2      sub {  
3          # first, clear the screen  
4          $app->draw_rect( [ 0, 0, $app->w, $app->h ], 0x000000FF );  
5  
6          # then render the ball  
7          $app->draw_rect( $ball->{rect}, 0xFF0000FF );  
8  
9          # ... and each paddle  
10         $app->draw_rect( $player1->{paddle}, 0xFF0000FF );  
11         $app->draw_rect( $player2->{paddle}, 0xFF0000FF );  
12  
13         # finally, update the screen  
14         $app->update;  
15     }  
16 );  
17  
18 # let's roll!  
19 $app->run;
```

This approach is rather simple. The code clears the screen by painting a black rectangle the size of the screen, then painting opaque red (0xFF0000FF) rectangles in each object's position.

The result can be seen on the screenshot:




Figure 6.1: First view of our Pong clone

6.3 Moving the Player's Paddle

It's time to let the player move the left paddle! Remember that motion is merely changing an object's position with respect to time. If this motion is, in the game, a magical teleportation, you can change the (x, y) coordinates and be done with it. If the motion needs to represent some sort of realistic physics, the object needs to move at an understood speed. Pong paddles have a constant speed, so there's no need to model acceleration. Also, as paddles move only vertically, the game only needs to track vertical velocity. Add a `v_y` element to each paddle structure:

```
my $player1 = {  
  paddle => SDLx::Rect->new( 10, $app->h / 2, 10, 40 ),  
  v_y    => 0,  
};
```

Chapter 6 | PONG!

Now what? How does this new attribute help modify the position of a paddle? Velocity represents the *displacement* how much displacement happens in a unit of time, as in 20 km/h or 4 m/s. In this Pong clone, the unit of time is the app's Δt . The velocity of a paddle is v_y pixels per Δt . Here is where the motion handlers come in handy:

```
# handles the player's paddle movement
$app->add_move_handler( sub {
    my ( $step, $app ) = @_;
    my $paddle = $player1->{paddle};
    my $v_y     = $player1->{v_y};

    $paddle->y( $paddle->y ( $v_y * $step ) );
});
```

If you recall previous chapters, the code should be straightforward. When v_y is 0 at any given run cycle, the paddle won't change its y position. If, however, there is a vertical velocity, the code updates the y position based on how much of the expected cycle time (the app's Δt) has passed. A value of 1 in $\$step$ indicates a full cycle has occurred, so that $\$v_y * \$step$ is the same as $\$v_y * 1$, which simplifies to $\$v_y$ – the desired speed for one cycle. If the handler gets called more frequently, the paddle will move a relatively shorter amount.

6.3.1 Rinse and Repeat

The second player's paddle will use the same motion mechanics, so it won't hurt to prepare for its motion:

```
1   my $player2 = {
2       paddle => SDLx::Rect->new( $app->w - 20, $app->h / 2, 10, 40),
3       v_y    => 0,
4   };
```

And add another motion handler, just like our player's:

```

1 # handles AI's paddle movement
2 $app->add_move_handler( sub {
3     my ( $step, $app ) = @_;
4     my $paddle = $player2->{paddle};
5     my $v_y     = $player2->{v_y};
6
7     $paddle->y( $paddle->y ( $v_y * $step ) );
8 });

```

For the sake of simplicity of explanation, this code has repetition a real program would not want. This repetition could go away in several ways. You could use an array to hold all moving elements. You could use a helper function to create a new closure for each paddle. You could turn the game object hash references into real objects and add a `move()` or `update_position()` method.

6.3.2 Move that Paddle!

Paddle velocity `v_y` has a default value of zero, so paddles begin by not moving. That's good, until the player wants to move the paddle. To divine the player's intent, the program must bind the up and down arrow keys of the keyboard to manipulate the positive and negative velocity of the paddle through an event hook. This means loading the `SDL::Events` module:

```

1 use SDL;
2 use SDL::Events;
3 use SDLx::App;
4 use SDLx::Rect;

```

... and creating an event hook:

```

1 # handles keyboard events
2 $app->add_event_handler(>
3     sub {

```

Chapter 6 | PONG!

```
4      my ( $event, $app ) = @_;
5
6      # user pressing a key
7      if ( $event->type == SDL_KEYDOWN ) {
8
9          # up arrow key means going up (negative velocity)
10         if ( $event->key_sym == SDLK_UP ) {
11             $player1->{v_y} = -2;
12         }
13
14         # down arrow key means going down (positive velocity)
15         elsif ( $event->key_sym == SDLK_DOWN ) {
16             $player1->{v_y} = 2;
17         }
18     }
19     # user releasing a key
20     elsif ( $event->type == SDL_KEYUP ) {
21
22         # up or down arrow keys released, stop the paddle
23         if (
24             $event->key_sym == SDLK_UP
25             or $event->key_sym == SDLK_DOWN
26         ) {
27             $player1->{v_y} = 0;
28         }
29     }
30 }
31 );
```

Again, there's nothing new. Whenever the user presses the up arrow key, the paddle should move up. Keep in mind that the origin point of 0, 0 in SDL is the top-left corner, so a negative `v_y` will decrease the paddle's `y` and send it **up** the screen. Similarly, adding a positive value to `v_y` whenever the user presses the down arrow key will move the paddle down. When the user releases either arrow key, assigning zero to `v_y` stops the motion.

6.4 A Bouncing Ball

The ball's movement is similar to that of either paddle, except that it also has a horizontal velocity component of `v_x`. Add that to the ball structure:

```
1  my $ball = {
2      rect => SDLx::Rect->new( $app->w / 2, $app->h / 2, 10, 10 ),
3      v_x  => -2.7,
4      v_y  => 1.8,
5  };
```

The ball will have an initial velocity of -2.7 horizontally and 1.8 vertically. Just as a negative vertical velocity moves the object up, a negative horizontal velocity moves it towards the left side of the screen. The ball also needs a motion handler to update its position according to its velocity:

```
1  handles the ball movement
2  app->add_move_handler( sub {
3      my ( $step, $app ) = @_;
4      my $ball_rect      = $ball->{rect};
5
6      $ball_rect->x( $ball_rect->x + ($ball->{v_x} * $step) );
7      $ball_rect->y( $ball_rect->y + ($ball->{v_y} * $step) );
8  });
```

All of these motion handlers look similar so far, but if you're paying close attention, you can probably spot a bug caused by missing code. Try running the game. You'll see the ball going, going, and gone!

This handler needs to confine the ball to the screen. Whenever the ball reaches a top or bottom edge of the screen, it needs to bounce. That's easy enough to add:

```
1  # handles the ball movement
2  $app->add_move_handler( sub {
3      my ( $step, $app ) = @_;
```

Chapter 6 | PONG!

```
4      my $ball_rect      = $ball->{rect};
5
6      $ball_rect->x( $ball_rect->x + ($ball->{v_x} * $step) );
7      $ball_rect->y( $ball_rect->y + ($ball->{v_y} * $step) );
8
9      # collision to the bottom of the screen
10     if ( $ball_rect->bottom >= $app->h ) {
11         $ball_rect->bottom( $app->h );
12         $ball->{v_y} *= -1;
13     }
14
15     # collision to the top of the screen
16     elsif ( $ball_rect->top <= 0 ) {
17         $ball_rect->top( 0 );
18         $ball->{v_y} *= -1;
19     }
20 });
```

If the new y ("bottom" or "top") value would take the ball off the screen in part or in whole, the handler updates the ball's position with the furthest position possible while remaining on the screen, so that the ball will only ever *touch* that edge. The handler also reverses *y* so that the ball will bounce back onto the screen going the opposite direction at the same speed.

6.4.1 He shoots... and scores!!

That fixes one bug, but what should happen when the ball hits the left or right edges of the screen? According to the rules of Pong, this means the player on the opposite side scored a point, and the ball should go back to the center of the screen. Start by adding a score attribute for each player:

```
1      my $player1 = {
2          paddle => SDLx::Rect->new( 10, $app->h / 2, 10, 40),
3          v_y    => 0,
4          score  => 0,
```

```

5     };
6
7     my $player2 = {
8         paddle => SDLx::Rect->new( $app->w - 20, $app->h / 2, 10, 40),
9         v_y     => 0,
10        score    => 0,
11    };

```

Then update the ball's motion handler to handle the out of bounds condition for the left and right borders:

```

1  # handles the ball movement
2  $app->add_move_handler( sub {
3      my ( $step, $app ) = @_ ;
4      my $ball_rect      = $ball->{rect};
5
6      $ball_rect->x( $ball_rect->x + ($ball->{v_x} * $step) );
7      $ball_rect->y( $ball_rect->y + ($ball->{v_y} * $step) );
8
9      # collision to the bottom of the screen
10     if ( $ball_rect->bottom >= $app->h ) {
11         $ball_rect->bottom( $app->h );
12         $ball->{v_y} *= -1;
13     }
14
15     # collision to the top of the screen
16     elsif ( $ball_rect->top <= 0 ) {
17         $ball_rect->top( 0 );
18         $ball->{v_y} *= -1;
19     }
20
21     # collision to the right: player 1 score!
22     elsif ( $ball_rect->right >= $app->w ) {
23         $player1->{score}++;
24         reset_game();
25         return;
26     }

```

```

27
28     # collision to the left: player 2 score!
29     elsif ( $ball_rect->left <= 0 ) {
30         $player2->{score}++;
31         reset_game();
32         return;
33     }
34 });

```

If the ball hits the right edge of the screen (the app's width), we increase player 1's score, call `reset_game()`, and return without updating the ball's position. If the ball hits the left edge of the screen, do the same for player 2.

The `reset_game()` function must return the ball to the center of the screen:

```

1  sub reset_game {
2      $ball->{rect}->x( $app->w / 2 );
3      $ball->{rect}->y( $app->h / 2 );
4  }

```

6.5 Collision Detection: The Ball and The Paddle

The game's existing collision detection is very simple because the paddles and ball can only collide with the fixed edges of the screen. The game gets more interesting when it can detect whether the ball and a paddle collide—or rather, intersect.

The Separating Axis Theorem roughly states that two convex shapes in a 2D plane *do not* intersect if and only you can place a line which separates them. Because the paddles and the ball are rectangular *and* aligned along one axis, detecting a collision means choosing one item and testing its top, right, bottom, and left lines for intersection. If any other object is on one side or the other of those four lines, there is no collision. Otherwise, there is a collision.

In more general terms, given two rects A and B, you can establish several conditions:




Figure 6.2: if B is completely to the left, right, top or bottom of A, they do NOT intersect

- if A's bottom side is above B's top side, then A is completely above B (fig. 6.2.1).
- if A's top side is below B's bottom side, then A is completely below B (fig. 6.2.2).
- if A's right side is to the left of B's left side, then A is completely to the left of B (fig. 6.2.3).
- if A's left side is to the right of B's right side, then A is completely to the right of B (fig 6.2.4).

Keep in mind that SDL's origin point of 0, 0 is always the top left corner. This produces a simple generic `check_collision()` function which returns true if two rect objects have collided:

```

1  sub check_collision {
2      my ($A, $B) = @_;
3
4      return if $A->bottom < $B->top;
5      return if $A->top    > $B->bottom;
6      return if $A->right  < $B->left;
7      return if $A->left   > $B->right;
8
9      # we have a collision!
10     return 1;
11 }

```

Chapter 6 | PONG!

The ball motion handler can now test to see if the ball has hit either paddle:

```
1  # handles the ball movement
2  $app->add_move_handler( sub {
3      my ( $step, $app ) = @_;
4      my $ball_rect = $ball->{rect};
5
6      $ball_rect->x( $ball_rect->x + ($ball->{v_x} * $step) );
7      $ball_rect->y( $ball_rect->y + ($ball->{v_y} * $step) );
8
9      # collision to the bottom of the screen
10     if ( $ball_rect->bottom >= $app->h ) {
11         $ball_rect->bottom( $app->h );
12         $ball->{v_y} *= -1;
13     }
14
15     # collision to the top of the screen
16     elsif ( $ball_rect->top <= 0 ) {
17         $ball_rect->top( 0 );
18         $ball->{v_y} *= -1;
19     }
20
21     # collision to the right: player 1 score!
22     elsif ( $ball_rect->right >= $app->w ) {
23         $player1->{score}++;
24         reset_game();
25         return;
26     }
27
28     # collision to the left: player 2 score!
29     elsif ( $ball_rect->left <= 0 ) {
30         $player2->{score}++;
31         reset_game();
32         return;
33     }
34
35     # collision with player1's paddle
36     elsif ( check_collision( $ball_rect, $player1->{paddle} ) ) {
```

```

37         $ball_rect->left( $player1->{paddle}->right );
38         $ball->{v_x} *= -1;
39     }
40
41     # collision with player2's paddle
42     elsif ( check_collision( $ball_rect, $player2->{paddle} )) {
43         $ball->{v_x} *= -1;
44         $ball_rect->right( $player2->{paddle}->left );
45     }
46 });

```

That's it! If the ball hits the first player's paddle, the handler reverses its horizontal velocity (`v_x`) to make it bounce back, and set its left edge to the paddle's right so they don't overlap. The logic is similar for the second player's paddle, except that the ball's right edge now must be at the same position as the paddle's left, as the ball has hit the other side of the paddle.

6.6 Artificial Stupidity

This Pong game is almost done. With scoring, ball movement, and paddle movement, it's playable—but dull, unless the second player can move. It's easy enough to bind a secondary set of keys to move the second paddle, but what if you want a quick game on your own without a friend around?

Artificial intelligence for games is a complex field of study, with many algorithms. Fortunately, the easiest approach is simple to model for Pong: the second player's paddle should follow the ball as it moves. All that takes is some new code in the second player's motion handler:

```

1     # handles AI's paddle movement
2     $app->add_move_handler( sub {
3         my ( $step, $app ) = @_;
4         my $paddle          = $player2->{paddle};
5         my $v_y              = $player2->{v_y};

```

Chapter 6 | PONG!

```
6
7     if ( $ball->{rect}->y > $paddle->y ) {
8         $player2->{v_y} = 1.5;
9     }
10    elseif ( $ball->{rect}->y < $paddle->y ) {
11        $player2->{v_y} = -1.5;
12    }
13    else {
14        $player2->{v_y} = 0;
15    }
16
17    $paddle->y( $paddle->y + ( $v_y * $step ) );
18 }
```

If the ball is below the paddle (if its y value is greater than the y value of the paddle), the paddle needs a positive velocity to go downwards. If, otherwise, the ball has a lower y value, the paddle's v_y gets a negative value. If the ball is somewhere in between those two values, the paddle stays in place.

6.7 Cosmetics: Displaying the Score

All that's left is polish. Displaying the score means drawing text to the screen. That's the purpose of the `SDLx::Text` module:

```
1 use SDL;
2 use SDL::Events;
3 use SDLx::App;
4 use SDLx::Rect;
5 use SDLx::Text;
```

Create an object to represent the display of the score:

```
1 my $score = SDLx::Text->new( font => 'font.ttf', h_align => 'center' );
```


The optional `font` parameter specifies the path to a TrueType Font. Feel free to change *font.ttf* as you like. Otherwise, leave out this parameter and SDL will use the bundled default font. The other parameter, `h_align`, allows you to specify the horizontal alignment of rendered text. The default is left alignment.

Add the score object to the show handler to display it:

```
1  $app->add_show_handler(  
2      sub {  
3          # first clear the screen  
4          $app->draw_rect( [0, 0, $app->w, $app->h], 0x000000FF );  
5  
6          # then render the ball  
7          $app->draw_rect( $ball->{rect}, 0xFF0000FF );  
8  
9          # ... and each paddle  
10         $app->draw_rect( $player1->{paddle}, 0xFF0000FF );  
11         $app->draw_rect( $player2->{paddle}, 0xFF0000FF );  
12  
13         # ... and each player's score!  
14         $score->write_to(  
15             $app,  
16             $player1->{score} . ' x ' . $player2->{score}  
17         );  
18  
19         # finally, update the screen  
20         $app->update;  
21     }  
22 );
```

The `write_to()` call will write to any surface passed as the first argument—in this case, the app itself. The second argument is the string to render. Note that the string's when rendered is relative to the surface to which it writes. The default position is (0, 0). Because the `$score` object has horizontal centering, the text will write to the top and center of the screen—not the top and left.

The result is:




Figure 6.3: our finished Pong clone, in all its glory

6.8 Exercises

Pong is a simple game, but there's plenty of room for polish. Here's your chance to add some features. Of course, there's always more than one way to do things:

- 1 Every time a player scores, the ball goes back to the middle but has the same sense and direction as before. See if you can make it restart at a random direction instead.
- 2 Red is boring. How about a psychedelic Pong? Pick three different colors and make each paddle oscillate between them every time the ball hits it.

6.8.1 Answers

- 1 To make the ball restart at a random direction, update `reset_game()` function to set the ball's `v_x` and `v_y` to a random value between. Anything between positive 1.5 and 2.5 works well:

```
1  sub reset_game {
2      $ball->{rect}->x( $app->w / 2 );
3      $ball->{rect}->y( $app->h / 2 );
4
5      $ball->{v_x} = (1.5 + int rand 1) * (rand 2 > 1 ? 1 : -1);
6      $ball->{v_y} = (1.5 + int rand 1) * (rand 2 > 1 ? 1 : -1);
7  }
```

- 2 Start by representing the available colors. You could use separate colors or hues for each player, but for simplicity this code uses a single group of colors. Each player's hash will contain the index into this array:

```
1  my @colors = qw( 0xFF0000FF 0x00FF00FF 0x0000FFFF 0xFFFF00FF );
2
3  my $player1 = {
4      paddle => SDLX::Rect->new( 10, $app->h / 2, 10, 40),
5      v_y    => 0,
6      score  => 0,
7      color  => 0,
8  };
9
10 my $player2 = {
11     paddle => SDLX::Rect->new( $app->w - 20, $app->h / 2, 10, 40),
12     v_y    => 0,
13     score  => 0,
14     color  => 0,
15 };
```

Now make the ball's color change every time a paddle hits it:

Chapter 6 | PONG!

```
1  # handles the ball movement
2  $app->add_move_handler( sub {
3
4      ...
5
6      # collision with player1's paddle
7      elsif ( check_collision( $ball_rect, $player1->{paddle} )) {
8          $ball_rect->left( $player1->{paddle}->right );
9          $ball->{v_x} *= -1;
10         $player1->{color} = ($player1->{color} + 1) % @colors;
11     }
12
13     # collision with player2's paddle
14     elsif ( check_collision( $ball_rect, $player2->{paddle} )) {
15         $ball->{v_x} *= -1;
16         $ball_rect->right( $player2->{paddle}->left );
17         $player2->{color} = ($player2->{color} + 1) % @colors;
18     }
19 });
```

Finally, change the show handler to use the current color referenced by `color`, instead of the previously hardcoded value:

```
1  $app->add_show_handler(
2      sub {
3          # first clear the screen
4          $app->draw_rect( [0, 0, $app->w, $app->h], 0x000000FF );
5
6          # then render the ball
7          $app->draw_rect( $ball->{rect}, 0xFF0000FF );
8
9          # ... and each paddle
10         $app->draw_rect( $player1->{paddle}, $colors[ $player1->{color} ] );
11         $app->draw_rect( $player2->{paddle}, $colors[ $player2->{color} ] );
12         ...
13
14         # finally update the screen
```

```
15         $app->update;  
16     }  
17 );
```

6.9 Author

This chapter's content graciously provided by Breno G. de Oliveira (garu).

6.10 POD ERRORS

Hey! **The above document had some coding errors, which are explained below:**

Around line 814:

You forgot a '=back' before '=head1'

7

Tetris

Pong is an important milestone in gaming history. If you can write it, you understand the basics of game programming. The next step in mastery comes from writing something like Tetris, with better animation and more complex scoring.

To follow along, download the sample code from https://github.com/PerlGameDev/SDL_Manual/raw/master/games/tetris.zip. To start the game, extract this Zip file and run:

```
$ perl tetris.pl
```




Figure 7.1: Tetris using SDLx Perl

7.1 The Game Window

The game starts out as you should expect by now:

```
1 use strict;  
2 use warnings;  
3  
4 use SDL;  
5 use SDL::Event;  
6 use SDL::Events;  
7 use SDLx::App;
```



```

8
9 # create the main screen
10 my $app = SDLx::App->new(
11     w          => 400,
12     h          => 512,
13     exit_on_quit => 1,
14     dt         => 0.2,
15     title      => 'SDLx Tetris'
16 );

```

This game requires several pieces of artwork, and so the program must manage and store them somehow. The `SDLx::Surface` module handles the conversion of files from their storage on disk into a format SDL can use, and an array will hold them:

```

1 use SDL;
2 use SDLx::Surface;
3 my $back = SDLx::Surface->load( 'data/tetris_back.png' );
4 my @piece = (undef);
5 push @piece, SDLx::Surface->load( "data/tetris_$.png" ) for 1..7;

```

The `$back` variable holds one special surface: the background image. Everything else is in the `@piece` array.

7.2 Managing Blocks

Blocks are critical to the success of a Tetris game. The program must represent them in a sensible way: they must be easy to access and they must be easy to manipulate and calculate. A hash fulfills the ease of access:

```

1 my %pieces = (
2     I => [0, 5, 0, 0,
3         0, 5, 0, 0,
4         0, 5, 0, 0,
5         0, 5, 0, 0],

```

Chapter 7 | TETRIS

```
6     J => [0, 0, 0, 0,
7           0, 0, 6, 0,
8           0, 0, 6, 0,
9           0, 6, 6, 0],
10    L => [0, 0, 0, 0,
11          0, 2, 0, 0,
12          0, 2, 0, 0,
13          0, 2, 2, 0],
14    O => [0, 0, 0, 0,
15          0, 3, 3, 0,
16          0, 3, 3, 0,
17          0, 0, 0, 0],
18    S => [0, 0, 0, 0,
19          0, 4, 4, 0,
20          4, 4, 0, 0,
21          0, 0, 0, 0],
22    T => [0, 0, 0, 0,
23          0, 7, 0, 0,
24          7, 7, 7, 0,
25          0, 0, 0, 0],
26    Z => [0, 0, 0, 0,
27          1, 1, 0, 0,
28          0, 1, 1, 0,
29          0, 0, 0, 0],
30 );
```

Each hash entry holds a four-element array reference which represents a grid of the piece. Each item in the array corresponds to an image in the `@piece` array. Drawing a piece means blitting one element of `@piece` for each non-zero entry in the piece's array.

```
1  use strict;
2  use warnings;
3
4  use List::Util qw(shuffle min max);
```

Selecting pieces needs some randomness. The core `List::Util` module can help:

```

1      Z => [0,0,0,0,
2            1,1,0,0,
3            0,1,1,0,
4            0,0,0,0],
5  );
6
7  my $next_tile      = shuffle keys %pieces;
8  my $curr_tile      = [ undef, 4, 0 ];
9  @{ $curr_tile->[0] } = @{ $pieces{$next_tile} };
10  $next_tile        = shuffle keys %pieces;

```

This code randomly chooses a `$next_tile`, then sets the piece data for the first piece in `$curr_tile`.

7.3 Piece Collisions

Collision detection is both easier (because only one piece at a time moves) and more difficult (because the screen continues to fill up with pieces). One solution is to treat the screen as two overlapping grids. The first grid represents the moving piece. The second grid represents the pieces already in place. When a moving piece collides with a piece in the fixed grid, the moving piece becomes stationary and joins the fixed grid. When that action clears one or more lines, the stationary grid changes.

Start by defining these grids:

```

1  push @piece, SDLx::Surface->load( "data/tetris_$_.png" ) for 1..7;
2
3  # compare the position of the moving piece with non-moving pieces
4  my $grid  = []; # moving piece
5  my $store = []; # non-moving pieces
6
7  my %pieces = (
8      I => [0,5,0,0,

```

Chapter 7 | TETRIS

Rotating a piece means transforming each of its elements:

```
1 sub rotate_piece {
2     my $_piece = shift;
3     my $_rotated = [];
4     my $_i = 0;
5
6     for (@$_piece) {
7         $_rotated->[ $_i + (($_i % 4 + 1) * 3)
8             - ( 5 * int( $_i / 4 )) ] = $_;
9         $_i++;
10    }
11
12    return $_rotated;
13 }
```

Collision detection requires checking both grids for a piece overlap in the direction the user wants to move the piece:

```
1 sub can_move_piece {
2     my $direction = shift;
3     my $amount = shift || 1;
4
5     for my $y (0 .. 3) {
6
7         for my $x (0 .. 3) {
8             if ($curr_tile->[0]->[ $x + 4 * $y ]) {
9                 return if $direction eq 'left'
10                    && $x - $amount + $curr_tile->[1] < 0;
11                 return if $direction eq 'right'
12                    && $x + $amount + $curr_tile->[1] > 9;
13                 return if $direction eq 'down'
14                    && int($y + $amount + $curr_tile->[2]) > 22;
15
16                 return if $direction eq 'right'
17                    && $store->[ $x + $amount +
18                        $curr_tile->[1] +
```

```

19             10 * int($y + $curr_tile->[2]) ];
20         return if $direction eq 'left'
21             && $store->[ $x - $amount      +
22                 $curr_tile->[1] +
23                 10 * int($y + $curr_tile->[2]) ];
24         return if $direction eq 'down'
25             && $store->[ $x +
26                 $curr_tile->[1]
27                 + 10 * int($y + $amount +
28                 $curr_tile->[2]) ];
29     }
30 }
31 }
32
33     return 1;
34 }

```

All of the pieces are in place to move the piece: make the collision check, then place the piece into the appropriate grid for its next position:

```

1  sub move_piece {
2      my $direction = shift;
3      my $amount     = shift || 1;
4
5      if ($direction eq 'right') {
6          $curr_tile->[1] += $amount;
7      }
8      elsif ($direction eq 'left') {
9          $curr_tile->[1] -= $amount;
10     }
11     elsif ($direction eq 'down') {
12         $curr_tile->[2] += $amount;
13     }
14
15     @{$sgrid} = ();
16
17     for my $y (0..3) {

```

Chapter 7 | TETRIS

```
18         for my $x (0..3) {
19             if ($curr_tile->[0]->[$x + 4 * $y]) {
20                 $grid->[ $x + $curr_tile->[1] +
21                     10 * ($y + int($curr_tile->[2])) ]
22                     = $curr_tile->[0]->[$x + 4 * $y];
23             }
24         }
25     }
26 }
27
28 sub store_piece {
29     for my $y (0..3) {
30         for my $x (0..3) {
31             if ($curr_tile->[0]->[$x + 4 * $y]) {
32                 $store->[ $x + $curr_tile->[1] + 10 *
33                     ($y + int($curr_tile->[2])) ]
34                     = $curr_tile->[0]->[$x + 4 * $y];
35             }
36         }
37     }
38 }
```

Of course this all needs an event handler to attempt to move the pieces appropriately:

```
1 sub trigger_move_event_handler {
2     my ( $event, $app ) = @_;
3
4     if ( $event->type == SDL_KEYDOWN ) {
5         my $key = $event->key_sym;
6
7         if ( $event->key_sym & (SDLK_LEFT|SDLK_RIGHT|SDLK_UP|SDLK_DOWN) ) {
8             if ($key == SDLK_LEFT && can_move_piece('left')) {
9                 move_piece('left');
10            }
11            elsif ($key == SDLK_RIGHT && can_move_piece('right')) {
12                move_piece('right');
13            }
14        }
15    }
16 }
```

```

14         elsif ($key == SDLK_DOWN && can_move_piece('down')) {
15             move_piece('down')
16         }
17         elsif ($key == SDLK_UP) {
18             $curr_tile->[0] = rotate_piece($curr_tile->[0]);
19         }
20     }
21 }
22 }
23
24 $app->add_event_handler( \&trigger_move_event_handler );

```

7.3.1 Score and Game State

The game state in Tetris is the combination of the fixed placement grid, the current piece, and the current score. The move handler can update all of these:

```

1 $app->add_move_handler( sub {
2     my ( $step, $app ) = @_;

```

Start by updating the current piece's state as movable or fixed:

```

1     if (can_move_piece('down', $step / 2)) {
2         # still movable
3         move_piece('down', $step / 2);
4     }
5     else {
6         # place the tile
7         store_piece($curr_tile);

```

Then update the state of the grid and check for lines to remove:

```

1         # checking for lines to delete
2         my $y;

```

Chapter 7 | TETRIS

```
3         my @to_delete);
4
5         for($y = 22; $y >= 0; $y--) {
6             # if the min value of this row is 0,
7             # it contains at least one open space
8             if (min( @{$store}[ ($y * 10)..((( $y + 1) *10 ) -1 )])) {
9                 push @to_delete, $y;
10            }
11        }
```

Deleting a line should increment the user's score:

```
1         # deleting lines
2         foreach (@to_delete) {
3             splice @{$store}, $_ * 10, 10;
4             $score++;
5         }
```

... and should clear that line off of the fixed grid:

```
1         # adding blank rows to the top
2         foreach (@to_delete) {
3             splice @{$store}, 0, 0, (0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0);
4         }
5
```

... and the game should launch a new tile.

```
1         # launching new tile
2         @{$curr_tile->[0]} = @{$pieces{$next_tile}};
3         $curr_tile->[1]    = 4;
4         $curr_tile->[2]    = 0;
5         $next_tile        = shuffle keys %pieces;
6     }
7 });
```


7.3.2 Drawing the Game

Those are the mechanics. How about displaying the game? The show handler needs to iterate through all of the elements in both grids and draw the appropriate tile:

```
1  $app->add_show_handler(  
2      sub {  
3          # first clear the screen  
4          $app->draw_rect( [ 0, 0, $app->w, $app->h ], 0x000000 );  
5  
6          # and draw the background image  
7          $back->blit( $app );  
8          my $x = 0;  
9          my $y = 0;  
10  
11         # draw the fixed tiles  
12         foreach (@{$store}) {  
13             $piece[$_]->blit( $app,  
14                             undef,  
15                             [ 28 + $_%10 * 20, 28 + $y * 20 ]  
16                             ) if $_;  
17             $x++;  
18             $y++ unless $x % 10;  
19         }  
20  
21         $x = 0;  
22         $y = 0;  
23  
24         # draw the moving tile  
25         foreach (@{$grid}) {  
26             $piece[$_]->blit( $app, undef,  
27                             [ 28 + $_ % 10 * 20, 28 + $y * 20 ] ) if $_;  
28             $x++;  
29             $y++ unless $x % 10;  
30         }  
31  
32         # the next tile will be...
```

Chapter 7 | TETRIS

```
33         my $next_tile_index = max( @{$pieces{$next_tile}} );
34     for $y (0..3) {
35         for $x (0..3) {
36             if ($pieces{$next_tile}->[$x + 4 * $y]) {
37                 $piece[$next_tile_index]->blit( $app, undef,
38                                                 [ 264 + $x * 20,
39                                                 48 + $y * 20 ]
40                                                 );
41             }
42         }
43     }
```

... and should draw the score:

```
1         $score_text->write_xy( $app, 248, 20, "Next Piece" );
2         $score_text->write_xy( $app, 248, 240, "Score: $score" );
3
4         # finally, update the screen
5         $app->update;
6     }
7 );
8
9 # all is set, run the app!
10 $app->run();
```

7.4 Author

Code for this chapter was provided by Tobias Leich “FROGGS”.

8

Puzz! A puzzle game

8.1 Abstract

We are now ready to write another complete game. Instead of listing the code and then explaining it, I will go through the process of how I might write it.

Puzz is a simple rearrangement puzzle. A random image from the folder `Puzz` is in is chosen and broken into a 4x4 grid. The top left corner piece is then taken away, and every other piece is then moved to a random position, scrambling the image up. The goal is then to move pieces which are in the 4 squares adjacent to the empty square on to the empty square, and eventually restore the image.




Figure 8.1: Credits to Sebastian Riedel (krai.h.com) for the Perl6 logo used with permission in the application.

8.2 The Window

So, first thing we do is create the window. I've decided I want each piece to be 100x100, so the window needs to be 400x400.

```
use strict;
use warnings;

use SDL;
use SDLx::App;

my $App = SDLx::App->new(w => 400, h => 400, t => 'Puzz');
```

Next thing we usually do is figure out what global vars we will be needing. As with \$App, I like to name my globals with title case, so they are easily distinguishable from lexical vars. The globals we need are the grid (the positions of the pieces), the images we have to use, the current image, and a construct that will give us piece movement, along with an animation.

```
my @Grid;  
my @Img;  
my $CurrentImg;  
my %Move;
```

For now, lets fill in @Grid with what it's going to look like:

```
@Grid = (  
    [0, 1, 2, 3],  
    [4, 5, 6, 7],  
    [8, 9, 10, 11],  
    [12, 13, 14, 15],  
);
```

0 will be our blank piece, but we could have chosen it to be any other number. When the grid looks like this, it's solved, so eventually we will need a way to scramble it. It's good enough for now, though.

8.3 Loading the images

To load the images, we would normally use `SDLx::Surface`, but we're going to do it the `libsdl` way with `SDL::Image` because we need to do our own error handling.

```
use SDL::Image;  
use SDL::GFX::Rotozoom 'SMOOTHING_ON';  
  
while(<./*>) {
```

Chapter 8 | PUZZ! A PUZZLE GAME

```
if(-f and my $i = SDL::Image::load($_)) {  
    $i = SDL::GFX::Rotozoom::surface_xy($i, 0, 400 / $i->w, 400 / $i->h, SMOOTHING_ON);  
    push @Img, $i;  
}  
else  
{  
    warn "Cannot Load $_: " . SDL::get_error() if $_ =~ /\.jpg|png|bmp/;  
}  
}  
$CurrentImg = $Img[rand @Img];  
  
die "Please place images in the Current Folder" if $#Img < 0;
```

We just go through every file in the current directory, and try to load it as an image. `SDL::Image::load` will return false if there was an error, so we want to discard it when that happens. If we used `SDLx::Surface` to load the images, we would get a warning every time a file fails to load as an image, which we don't want. The `my $i = SDL::Image::load($_)` is just an idiom for setting a var and checking it for truth at the same time.

We want the image to be 400x400, and `SDL::GFX::Rotozoom` makes this possible. The two Rotozoom functions that are the most useful are `surface` and `surface_xy`. They work like this:

```
$zoomed_src = SDL::GFX::Rotozoom::surface($src, $angle, $zoom, $smoothing)  
$zoomed_src = SDL::GFX::Rotozoom::surface_xy($src, $angle, $x_zoom, $y_zoom, $smoothing)
```

The zoom values are the multiplier for that component, or for both components at once as with `$zoom`. `$angle` is an angle of rotation in degrees. `$smoothing` should be `SMOOTHING_ON` or `SMOOTHING_OFF` (which can be exported by `SDL::GFX::Rotozoom`) or just 1 or 0.

Once the image is zoomed, it is added to the image array. The current image is then set to a random value of the array.

8.4 Handling Events

The next part I like to write is the events. We're going to make Escape quit, and left click will move the pieces around. We use `SDL::Events` for the constants.

```
use SDL::Events;

sub on_event {
    my ($e) = @_;
    if($e->type == SDL_QUIT or $e->type == SDL_KEYDOWN and $e->key_sym == SDLK_ESCAPE) {
        $App->stop;
    }
    elsif($e->type == SDL_MOUSEBUTTONDOWN and $e->button_button == SDL_BUTTON_LEFT) {
        ...
    }
}

$App->add_event_handler(\&on_event);
# $App->add_move_handler(\&on_move);
# $App->add_show_handler(\&on_show);
$App->run;
```

8.5 Filling the Grid

Once we have something like this, it's a good time to put some warn messages in to make sure the inputs are working correctly. Once they are, it's time to fill it in.

```
my $x = int($e->button_x / 100);
my $y = int($e->button_y / 100);
if(!%Move and $Grid[$y][$x]) {`
    ...
}
```

From the pixel coordinates of the click (0 to 399), we want to find out the grid coordinates (0 to 3), so we divide both components by 100 and round them down. Then, we only want to continue on to see if that piece can move if no other piece is moving (%Move is false), and the piece clicked isn't the blank piece (0).

```
for([-1, 0], [0, -1], [1, 0], [0, 1]) {
    my $nx = $x + $_->[0];
    my $ny = $y + $_->[1];
    if($nx >= 0 and $nx < 4 and $ny >= 0 and $ny < 4 and !$Grid[$ny][$nx]) {
        ...
    }
}
```

8.6 Moving the Pieces

We check that the blank piece is in the 4 surrounding places by constructing 4 vectors. These will take us to those squares. The x component is first and the second is y. We iterate through them, setting \$nx and \$ny to the new position. Then if both \$nx and \$ny are within the grid (0 to 3), and that position in the grid is 0, we can move the piece to the blank square.

```
%Move = (
    x      => $x,
    y      => $y,
    x_dir  => $_->[0],
    y_dir  => $_->[1],
    offset => 0,
);
```

To make a piece move, we construct the move hash with all the information it needs to move the piece. The x and y positions of the piece, the x and y directions it will be moving (the vector), and it's current pixel offset from it's position (for the moving animation), which starts at 0.

8.6.1 The Move Handler Callback

Next we will write the move handler. All it needs to do is move any moving piece along by updating the offset, and click it in to where it's being moved to when it has moved the whole way (offset is 100 or more).

```
sub on_move {
  if(%Move) {
    $Move{offset} += 30 * $_[0];
    if($Move{offset} >= 100) {
      $Grid[$Move{y} + $Move{y_dir}][$Move{x} + $Move{x_dir}] = $Grid[$Move{y}][$Move{x}];
      $Grid[$Move{y}][$Move{x}] = 0;
      undef %Move;
    }
  }
}
```

30 has been arbitrarily chosen as the speed of the move, as it felt the best after a little playing and tweaking. Always remember to multiply things like this by the step value in `$_[0]` so that the animation moves in correct time with the updating.

Once the offset is 100 or more, the grid place that the piece is moving to is set to the value of the piece, and the piece is set to the blank value. The move is then finished, so `%Move` is deleted.

8.7 Rendering the Game

Now that we have all the functionality we need it's finally time to see the game.

```
sub on_show {
  $App->draw_rect( [0,0,$App->w,$App->h], 0 );
  for my $y (0..3) {
```

Chapter 8 | PUZZ! A PUZZLE GAME

```
        for my $x (0..3) {  
            ...  
        }  
    }  
    $App->flip;  
}
```

We start the show handler by drawing a black rect over the entire app. Entire surface and black are the defaults of `draw_rect`, so letting it use the defaults is good. Next we iterate through a `y` and `x` of 0 to 3 so that we can go through each piece of the grid. At the end of the handler we update the app with a call to `flip`.

```
next unless my $val = $Grid[$y][$x];  
my $xval = $val % 4;  
my $yval = int($val / 4);  
my $move = %Move && $Move{x} == $x && $Move{y} == $y;  
...  

```

Inside the two loops we put this. First we set `$val` to the grid value at the current position, and we skip to the next piece if it's the blank piece. We have the `x` and `y` coordinates of where that piece is on the board, but we need to figure out where it is on the image. If you refer back to the initialisation of the grid, the two operations to find the values should make sense. `$move` is set with a bool of whether it is this piece that is moving, if there is a piece moving at all.

```
$App->blit_by(  
    $CurrentImg,  
    [$xval * 100, $yval * 100, 100, 100],  
    [$x * 100 + ($move ? $Move{offset} * $Move{x_dir} : 0),  
     $y * 100 + ($move ? $Move{offset} * $Move{y_dir} : 0)]  
);
```

Now that we have all of this, we can blit the portion of the current image we need to the app. We use `blit_by` because the image we're blitting isn't an `SDLx::Surface` (because we didn't load it as one), but the app is. Here's how `blit_by` works as opposed to `blit`:

```

$src->blit($dest, $src_rect, $dest_rect)
$dest->blit_by($src, $src_rect, $dest_rect)

```

The portion we need is from the `$xval` and `$yval`, and where it needs to go to is from `$x` and `$y`. All are multiplied by 100 because we're dealing with 0 to 300, not 0 to 3. If the piece is moving, the offset multiplied by the direction is added to the position.

When the code is run with all 3 handlers, we have a fully working game. The pieces move around nicely when clicked. The only things it still needs are a shuffled grid and a way to check if the player has won. To implement these two things, we will make two more functions.

```

use List::Util 'shuffle';

sub new_grid {
    my @new = shuffle(0..15);
    @Grid = map { [ @new[ $_*4..$_*4+3 ] ] } 0..3;
    $CurrentImg = $Img[rand @Img];
}

```

We will replace the grid initialising we did with this sub. First it shuffles the numbers 0 through 15 with `List::Util::shuffle`. This array is then arranged into a 2D grid with a `map` and put in to `@Grid`. Setting the current image is also put into this sub.

```

sub won {
    my $correct = 0;
    for(@Grid) {
        for(@$_) {
            return 0 if $correct != $_;
            $correct++;
        }
    }
    return 1;
}

```

This sub returns whether the grid is in the winning configuration, that is, all piece values are in order from 0 to 15.

Now we put a call to `new_grid` to replace the grid initialisation we had before. We put `won` into the event handler to make click call `new_grid` if you have won. Finally, `won` is put into the show handler to show the blank piece if you have won.

8.8 Complete Code

Here is the finished code:

```

1  use strict;
2  use warnings;
3
4  use SDL;
5  use SDLx::App;
6  use SDL::Events;
7  use SDL::Image;
8  use SDL::GFX::Rotozoom 'SMOOTHING_ON';
9  use List::Util 'shuffle';
10
11 my $App = SDLx::App->new(w => 400, h => 400, t => 'Puzz');
12
13 my @Grid;
14 my @Img;
15 my $CurrentImg;
16 my %Move;
17
18 while(<./*>) {
19     if(-f and my $i = SDL::Image::load($_)) {
20         $i = SDL::GFX::Rotozoom::surface_xy($i, 0, 400 / $i->w, 400 / $i->h, SMOOTHING_ON);
21         push @Img, $i;
22     }
23     else

```

```

24     {
25         warn "Cannot Load $_: " . SDL::get_error() if $_ =~ /jpg|png|bmp/;
26     }
27
28 }
29
30 die "Please place images in the Current Folder" if $#Img < 0;
31
32 new_grid();
33
34 sub on_event {
35     my ($e) = @_;
36     if($e->type == SDL_QUIT or $e->type == SDL_KEYDOWN and $e->key_sym == SDLK_ESCAPE) {
37         $App->stop;
38     }
39     elsif($e->type == SDL_MOUSEBUTTONDOWN and $e->button_button == SDL_BUTTON_LEFT) {
40         my($x, $y) = map { int($_ / 100) } $e->button_x, $e->button_y;
41         if(won()) {
42             new_grid();
43         }
44         elsif(!%Move and $Grid[$y][$x]) {
45             for([-1, 0], [0, -1], [1, 0], [0, 1]) {
46                 my($nx, $ny) = ($x + $_->[0], $y + $_->[1]);
47                 if($nx >= 0 and $nx < 4 and $ny >= 0 and $ny < 4 and !$Grid[$ny][$nx]) {
48                     %Move = (
49                         x      => $x,
50                         y      => $y,
51                         x_dir  => $_->[0],
52                         y_dir  => $_->[1],
53                         offset => 0,
54
55                     );
56                 }
57             }
58         }
59     }
60
61     sub on_move {

```

Chapter 8 | PUZZ! A PUZZLE GAME

```
62     if(%Move) {
63         $Move{offset} += 30 * $_[0];
64         if($Move{offset} >= 100) {
65             $Grid[$Move{y} + $Move{y_dir}][$Move{x} + $Move{x_dir}] = $Grid[$Move{y}][$Move{x}];
66             $Grid[$Move{y}][$Move{x}] = 0;
67             undef %Move;
68         }
69     }
70 }
71
72 sub on_show {
73     $App->draw_rect( [0,0,$App->w,$App->h], 0 );
74     for my $y (0..3) {
75         for my $x (0..3) {
76             next if not my $val = $Grid[$y][$x] and !won();
77             my $xval = $val % 4;
78             my $yval = int($val / 4);
79             my $move = %Move && $Move{x} == $x && $Move{y} == $y;
80             $App->blit_by(
81                 $CurrentImg,
82                 [$xval * 100, $yval * 100, 100, 100],
83                 [$x * 100 + ($move ? $Move{offset} * $Move{x_dir} : 0),
84                 $y * 100 + ($move ? $Move{offset} * $Move{y_dir} : 0)]
85             );
86         }
87     }
88     $App->flip;
89 }
90
91 sub new_grid {
92     my @new = shuffle(0..15);
93     @Grid = map { [@new[ $_*4..$_*4+3 ]] } 0..3;
94     $CurrentImg = $Img[rand @Img];
95 }
96
97 sub won {
98     my $correct = 0;
99     for(@Grid) {
```

```

100         for(@$_) {
101             return 0 if $correct != $_;
102             $correct++;
103         }
104     }
105     return 1;
106 }
107
108 $App->add_event_handler(\&on_event);
109 $App->add_move_handler(\&on_move);
110 $App->add_show_handler(\&on_show);
111 $App->run;

```

You now hopefully know more of the process that goes in to creating a simple game. The process of creating a complex game is similar, it just requires more careful planning. You should have also picked up a few other tricks, like with `SDL::GFX::Rotozoom`, `SDL::Image::load` and `blit_by`.

8.9 Activities

1. Make the blank piece the bottom right piece instead of the top left piece.
2. Make the grid dimensions variable by getting the value from `$ARGV[0]`. The grid will then be 5x5 if `$ARGV[0]` is 5 and so on.

8.10 Author

This chapter's content graciously provided by Blaizer.

9

Sound and Music

Sound and Music in SDL are handled by the `Audio` and `SDL_Mixer` components. Enabling `Audio` devices is provided with the Core SDL Library and only supports wav files. `SDL_Mixer` supports more audio file formats and has additional features that we need for sound in Game Development.

Similarly to video in SDL, there are several way for perl developers to access the Sound components of SDL. For the plain `Audio` component the `SDL::Audio` and related modules are available. `SDL_Mixer` is supported with th `SDL::Mixer` module. There is currently a `SDLx::Sound` module in the work, but not completed at the time of writing this manual. For that reason this chapter will use `SDL::Audio` and `SDL::Mixer`.

9.1 Simple Sound Script

To begin using sound we must enable and open an audiospec:

```
se strict;
se warnings;
se SDL;
se Carp;
se SDL::Audio;
se SDL::Mixer;

SDL::init(SDL_INIT_AUDIO);

unless( SDL::Mixer::open_audio( 44100, AUDIO_S16SYS, 2, 4096 ) == 0 )

    Carp::croak "Cannot open audio: ".SDL::get_error();
```

`open_audio` will open an audio device with frequency at 44100 Mhz, audio format `AUDIO_S16SYS` (Note: This is currently the most portable format, however there are others), 2 channels and a chunk size of 4096. Fiddle with these values if you are comfortable with sound terminology and techniques.

9.1.1 Loading Samples

Next we will load sound samples that generally used for sound effects and the like. Currently `SDL::Mixer` reserves samples for `.WAV`, `.AIFF`, `.RIFF`, `.OGG`, and `.VOC` formats.

Samples run on one of the 2 channels that we opened up, while the other channel will be reserved for multiple plays of the sample. To load samples we will be doing the following:

```

use SDL::Mixer::Samples;

#Brilliant Lazer Sound from HTTP://FreeSound.Org/samplesViewSingle.php?id=30935
my $sample = SDL::Mixer::Samples::load_WAV('data/sample.wav');

unless($sample)
{
    Carp::croak "Cannot load file data/sample.wav: ".SDL::get_error();
}

```

9.1.2 Playing the sample and closing audio

Now we can play that sample on any open channel looping forever:

```

use SDL::Mixer::Samples;
use SDL::Mixer::Channels;

my $sample = SDL::Mixer::Samples::load_WAV('data/sample.wav');
unless( $sample)

    Carp::croak "Cannot load file data/sample.wav: ".SDL::get_error();

my $playing_channel = SDL::Mixer::Channels::play_channel( -1, $sample, 0 );

```

`play_channel` allows us to assign a sample to the channel `-1` which indicates any open channel. `0` indicates we want to play the sample only once.

Note that since the sound will be playing in an external process we will need to keep the perl script running. In a game this is no problem but for a single script like this we can just use a simple sleep function. Once we are done we can go ahead and close the audio device.

```

sleep(1);
SDL::Mixer::close_audio();

```

9.1.3 Streaming Music

Next we will use `SDL::Mixer::Music` to add a background music to our script here.

```
use SDL::Mixer::Channels;
+use SDL::Mixer::Music;

+#Load our awesome music from HTTP://8BitCollective.Com
+my $background_music =
+    SDL::Mixer::Music::load_MUS('data/music/01-PC-Speaker-Sorrow.ogg');

+unless( $background_music )
+{
+    Carp::croak "Cannot load music file data/music/01-PC-Speaker-Sorrow.ogg: ".SDL::get_error();
+}
```

Music types in `SDL::Mixer` run in a separate channel from our samples which allows us to have sound effects (like jump, or lasers etc) to play at the same time.

```
SDL::Mixer::Music::play_music($background_music,0);
```

`play_music` also takes a parameter for how many loops you would like to play the song for, where 0 is 1.

To stop the music we can call `halt_music`.

```
sleep(2);
SDL::Mixer::Music::halt_music();
SDL::Mixer::close_audio();
```

Controlling Volume can be as simple as:

```
All channels indicated by the -1
DL::Mixer::Channels::volume(-1,10);
```

```
Specifically for the Music
DL::Mixer::Music::volume_music( 10 );
```

Volumes can be set at anytime and range from 1-100.

9.1.4 Code so far

```
1  use strict;
2  use warnings;
3  use SDL;
4  use Carp;
5  use SDL::Audio;
6  use SDL::Mixer;
7  use SDL::Mixer::Samples;
8  use SDL::Mixer::Channels;
9  use SDL::Mixer::Music;
10 SDL::init(SDL_INIT_AUDIO);
11
12 unless( SDL::Mixer::open_audio( 44100, AUDIO_S16SYS, 2, 4096 ) == 0 )
13 {
14     Carp::croak "Cannot open audio: ".SDL::get_error();
15 }
16
17
18 my $sample = SDL::Mixer::Samples::load_WAV('data/sample.wav');
19
20 unless( $sample)
21 {
22     Carp::croak "Cannot load file data/sample.wav: ".SDL::get_error();
23 }
24
```

Chapter 9 | SOUND AND MUSIC

```
25 my $playing_channel = SDL::Mixer::Channels::play_channel( -1, $sample, 0 );
26
27 #Load our awesome music from HTTP://8BitCollective.Com
28 my $background_music = SDL::Mixer::Music::load_MUS('data/music/01-PC-Speaker-Sorrow.ogg');
29
30 unless( $background_music )
31 {
32     Carp::croak "Cannot load music file data/music/01-PC-Speaker-Sorrow.ogg: "
33         .SDL::get_error();
34 }
35
36 SDL::Mixer::Music::play_music( $background_music,0 );
37
38 sleep(2);
39
40 SDL::Mixer::Music::halt_music();
41 SDL::Mixer::close_audio;
```

9.2 Sound Applications

Now that we know how to prepare and play simple sounds we will apply it to an `SDLx::App`.

9.2.1 SDLx::App Audio Initialization

`SDLx::App` will initialize everything normally for us. However for a stream line application it is recommend to initialize only the things we need. In this case that is `SDL_INIT_VIDEO` and `SDL_INIT_AUDIO`.

```
use strict;
use warnings;
use SDL;
```

```

use Carp;
use SDLx::App;
use SDL::Audio;
use SDL::Mixer;
use SDL::Event;
use SDL::Events;
use SDL::Mixer::Music;
use SDL::Mixer::Samples;
use SDL::Mixer::Channels;

my $app = SDLx::App->new(
    init => SDL_INIT_AUDIO | SDL_INIT_VIDEO,
    width => 250,
    height => 75,
    title => "Sound Event Demo",
    eoq   => 1

```

9.2.2 Loading Resources

It is highly recommended to perform all resource allocations before a `SDLx::App::run()` method is called.

```

# Initialize the Audio
unless ( SDL::Mixer::open_audio( 44100, AUDIO_S16SYS, 2, 4096 ) == 0 ) {
    Carp::croak "Cannot open audio: " . SDL::get_error();
}

#Something to show while we play music and sounds
my $channel_volume = 100;
my $music_volume   = 100;
my $laser_status   = 'none';
my $music_status    = 'not playing';

# Load our sound resources

```

Chapter 9 | SOUND AND MUSIC

```
my $laser = SDL::Mixer::Samples::load_WAV('data/sample.wav');
unless ($laser) {
    Carp::croak "Cannot load sound: " . SDL::get_error();
}

my $background_music =
SDL::Mixer::Music::load_MUS('data/music/01-PC-Speaker-Sorrow.ogg');
unless ($background_music) {
    Carp::croak "Cannot load music: " . SDL::get_error();
}
```

9.2.3 The Show Handler

For the purposes of describing the current state of the music lets draw text to the screen in a show_handler.

```
$app->add_show_handler(
sub {

    $app->draw_rect([0,0,$app->w,$app->h], 0 );

    $app->draw_gfx_text( [10,10], [255,0,0,255], "Channel Volume : $channel_volume" );
    $app->draw_gfx_text( [10,25], [255,0,0,255], "Music Volume : $music_volume" );
    $app->draw_gfx_text( [10,40], [255,0,0,255], "Laser Status : $laser_status" );
    $app->draw_gfx_text( [10,55], [255,0,0,255], "Music Status : $music_status" );

    $app->update();

}
);
```

This will draw the channel volume of our samples, and the volume of the music. It will also print the status of our two sounds in the application.

9.2.4 The Event Handler

Finally our event handler will do the actual leg work and trigger the music and sound as we need it.

```
$app->add_event_handler(  
  sub {  
    my $event = shift;  
  
    if ( $event->type == SDL_KEYDOWN ) {  
      my $keysym = $event->key_sym;  
      my $keyname = SDL::Events::get_key_name($keysym);  
  
      if ( $keyname eq 'space' ) {  
  
        $laser_status = 'PEW!';  
        #fire lasers!  
        SDL::Mixer::Channels::play_channel( -1, $laser, 0 );  
  
      }  
      elsif ( $keyname eq 'up' ) {  
        $channel_volume += 5 unless $channel_volume == 100;  
      }  
      elsif ( $keyname eq 'down' ) {  
        $channel_volume -= 5 unless $channel_volume == 0;  
      }  
      elsif ( $keyname eq 'right' ) {  
        $music_volume += 5 unless $music_volume == 100;  
      }  
      elsif ( $keyname eq 'left' ) {  
        $music_volume -= 5 unless $music_volume == 0;  
      }  
      elsif ( $keyname eq 'return' ) {  
        my $playing = SDL::Mixer::Music::playing_music();  
        my $paused = SDL::Mixer::Music::paused_music();  
  
        if ( $playing == 0 && $paused == 0 ) {
```

Chapter 9 | SOUND AND MUSIC

```
        SDL::Mixer::Music::play_music( $background_music, 1 );
        $music_status = 'playing';
    }
    elsif ( $playing && !$paused ) {
        SDL::Mixer::Music::pause_music();
        $music_status = 'paused'
    }
    elsif ( $playing && $paused ) {
        SDL::Mixer::Music::resume_music();
        $music_status = 'resumed playing';
    }
}

SDL::Mixer::Channels::volume( -1, $channel_volume );
SDL::Mixer::Music::volume_music($music_volume);

}

}

);
```

The above event handler fires the laser on pressing the 'Space' key. Go ahead and press it multiple times as if you are firing a gun in a game! You will notice that depending on how fast you fire the laser the application will still manage to overlap the sounds as needed. The sample overlapping is accomplished by requiring multiple channels in the `open.audio` call. If your game has lots of samples that may play at the same time you may need more channels allocated. Additionally you can see that the volume control is easily managed both on the channels and the music with just incrementing or decrementing a value and calling the appropriate function.

Finally it is worth noticing the various state the background music can be in.

Lets run this application and the make sure to clean up the audio on the way out. `$app->run(); SDL::Mixer::Music::halt_music(); SDL::Mixer::close_audio;`

9.2.5 Completed Code

```
1  use strict;
2  use warnings;
3
4  use Cwd;
5  use Carp;
6  use File::Spec;
7
8  use threads;
9  use threads::shared;
10
11 use SDL;
12 use SDL::Event;
13 use SDL::Events;
14
15 use SDL::Audio;
16 use SDL::Mixer;
17 use SDL::Mixer::Music;
18 use SDL::Mixer::Effects;
19
20 use SDLx::App;
21 my $app = SDLx::App->new(
22     init    => SDL_INIT_AUDIO | SDL_INIT_VIDEO,
23     width   => 800,
24     height  => 600,
25     depth   => 32,
26     title   => "Music Visualizer",
27     eoq      => 1,
28     dt      => 0.2,
29 );
30
31 # Initialize the Audio
32 unless ( SDL::Mixer::open_audio( 44100, AUDIO_S16, 2, 1024 ) == 0 ) {
33     Carp::croak "Cannot open audio: " . SDL::get_error();
34 }
35
```

Chapter 9 | SOUND AND MUSIC

```
36 # Load our music files
37 my $data_dir = '.';
38 my @songs    = glob 'data/music/*.ogg';
39
40 my @stream_data : shared;
41
42 # Music Effect to pull Stream Data
43 sub music_data {
44     my ( $channel, $samples, $position, @stream ) = @_;
45
46     {
47         lock(@stream_data);
48         push @stream_data, @stream;
49     }
50
51     return @stream;
52 }
53
54 sub done_music_data { }
55
56 my $music_data_effect_id =
57     SDL::Mixer::Effects::register( MIX_CHANNEL_POST, "main::music_data",
58     "main::done_music_data", 0 );
59
60 # Music Playing Callbacks
61 my $current_song = 0;
62 my $lines = $ARGV[0] || 50;
63
64 my $current_music_callback = sub {
65     my ( $delta, $app ) = @_;
66
67     $app->draw_rect( [ 0, 0, $app->w(), $app->h() ], 0x000000FF );
68     $app->draw_gfx_text(
69         [ 5, $app->h() - 10 ],
70         [ 255, 0, 0, 255 ],
71         "Playing Song: " . $songs[ $current_song - 1 ]
72     );
73 }
```

```

74     my @stream;
75     {
76         lock @stream_data;
77         @stream      = @stream_data;
78         @stream_data = ();
79     }
80
81     # To show the right amount of lines we choose a cut of the stream
82     # this is purely for asthetic reasons.
83
84     my $cut = @stream / $lines;
85
86     # The width of each line is calculated to use.
87     my $l_wdt = ( $app->w() / $lines ) / 2;
88
89     for ( my $i = 0 ; $i < $#stream ; $i += $cut ) {
90
91         # In stereo mode the stream is split between two alternating streams
92         my $left  = $stream[$i];
93         my $right = $stream[ $i + 1 ];
94
95         # For each bar we calculate a Y point and a X point
96         my $point_y = ( ( ($left) ) * $app->h() / 4 / 32000 ) + ( $app->h / 2 );
97         my $point_y_r =
98             ( ( ($right) ) * $app->h() / 4 / 32000 ) + ( $app->h / 2 );
99         my $point_x = ( $i / @stream ) * $app->w;
100
101         # Using the parameters
102         # Surface, box coordinates and color as RGBA
103         SDL::GFX::Primitives::box_RGBA(
104             $app,
105             $point_x - $l_wdt,
106             $app->h() / 2,
107             $point_x + $l_wdt,
108             $point_y, 40, 0, 255, 128
109         );
110         SDL::GFX::Primitives::box_RGBA(
111             $app,

```

Chapter 9 | SOUND AND MUSIC

```
112         $point_x - $l_wdt,
113         $app->h() / 2,
114         $point_x + $l_wdt,
115         $point_y_r, 255, 0, 40, 128
116     );
117
118 }
119
120 $app->flip();
121
122 };
123
124 my $cms_move_callback_id;
125 my $pns_move_callback_id;
126 my $play_next_song_callback;
127
128 sub music_finished_playing {
129     SDL::Mixer::Music::halt_music();
130
131     $pns_move_callback_id = $app->add_move_handler($play_next_song_callback)
132     if ( defined $play_next_song_callback );
133
134 }
135
136 $play_next_song_callback = sub {
137     return $app->stop() if $current_song >= @songs;
138     my $song = SDL::Mixer::Music::load_MUS( $songs[ $current_song++ ] );
139     SDL::Mixer::Music::play_music( $song, 0 );
140
141     $app->remove_move_handler($pns_move_callback_id)
142     if defined $pns_move_callback_id;
143 };
144
145 $app->add_show_handler($current_music_callback);
146 $pns_move_callback_id = $app->add_move_handler($play_next_song_callback);
147
148 $app->add_move_handler(
149     sub {
```

```

150         my $music_playing = SDL::Mixer::Music::playing_music();
151
152         music_finished_playing() unless $music_playing;
153
154     }
155 );
156
157 $app->add_event_handler(
158     sub {
159         my ( $event, $app ) = @_;
160         if ( $event->type == SDL_KEYDOWN && $event->key_sym == SDLK_DOWN ) {
161
162             # Indicate that we are done playing the music_finished_playing
163             music_finished_playing();
164         }
165     }
166 );
167
168 $app->run();
169
170 SDL::Mixer::Effects::unregister( MIX_CHANNEL_POST, $music_data_effect_id );
171 SDL::Mixer::Music::hook_music_finished();
172 SDL::Mixer::Music::halt_music();
173 SDL::Mixer::close_audio();

```

9.3 Music Visualizer

The music visualizer example processes real-time sound data—data as it plays—and displays the wave form on the screen. It will look something like:




Figure 9.1: Simple Music Visualization

9.3.1 The Code and Comments

The program begins with the usual boilerplate of an SDL Perl application:

```
use strict;
use warnings;

use Cwd;
use Carp;
use File::Spec;

use threads;
use threads::shared;

use SDL;
use SDL::Event;
use SDL::Events;

use SDL::Audio;
use SDL::Mixer;
use SDL::Mixer::Music;
```



```
use SDL::Mixer::Effects;
```

```
use SDL::App;
```

It then creates an application with both audio and video support:

```
my $app = SDL::App->new(  
    init    => SDL_INIT_AUDIO | SDL_INIT_VIDEO,  
    width   => 800,  
    height  => 600,  
    depth   => 32,  
    title   => "Sound Event Demo",  
    eoq     => 1,  
    dt      => 0.2,  
);
```

The application must initialize the audio system with a format matching the expected audio input. `AUDIO_S16` provides a 16-bit signed integer array for the stream data:

```
# Initialize the Audio  
unless ( SDL::Mixer::open_audio( 44100, AUDIO_S16, 2, 1024 ) == 0 ) {  
    Carp::croak "Cannot open audio: " . SDL::get_error();  
}
```

The music player needs the music files from the *data/music/* directory:

```
# Load our music files  
my $data_dir = '.';  
my @songs = glob 'data/music/*.ogg';
```

A music effect reads stream data, then serializes it to share between threads:

```
my @stream_data : shared;  
  
# Music Effect to pull Stream Data  
sub music_data {
```

Chapter 9 | SOUND AND MUSIC

```
my ( $channel, $samples, $position, @stream ) = @_;

{
    lock(@stream_data);
    push @stream_data, @stream;
}

return @stream;
}

sub done_music_data { }
```

... and that effect gets registered as a callback with `SDL::Mixer::Effects`:

```
my $music_data_effect_id =
    SDL::Mixer::Effects::register( MIX_CHANNEL_POST, "main::music_data",
                                   "main::done_music_data", 0 );
```

The program's single command-line option governs the number of lines to display in the visualizer. The default is 50.

```
my $lines = $ARGV[0] || 50;
```

The drawing callback for the `SDLx::App` runs while a song plays. It reads the stream data and displays it on the screen as a wave form. The math behind calculating the graphics to display is more detail than this article intends, but the graphic code is straightforward:

```
# Music Playing Callbacks
my $current_song = 0;

my $current_music_callback = sub {
    my ( $delta, $app ) = @_;

    $app->draw_rect( [ 0, 0, $app->w(), $app->h() ], 0x000000FF );
    $app->draw_gfx_text(
        [ 5, $app->h() - 10 ],
```

```

        [ 255, 0, 0, 255 ],
        "Playing Song: " . $songs[ $current_song - 1 ]
    );

my @stream;
{
    lock @stream_data;
    @stream      = @stream_data;
    @stream_data = ();
}

# To show the right amount of lines we choose a cut of the stream
# this is purely for asthetic reasons.

my $cut = @stream / $lines;

# The width of each line is calculated to use.
my $l_wdt = ( $app->w() / $lines ) / 2;

for ( my $i = 0 ; $i < $#stream ; $i += $cut ) {

    # In stereo mode the stream is split between two alternating streams
    my $left  = $stream[$i];
    my $right = $stream[ $i + 1 ];

    # For each bar we calculate a Y point and a X point
    my $point_y = ( ( ($left) ) * $app->h() / 4 / 32000 ) + ( $app->h / 2 );
    my $point_y_r =
        ( ( ($right) ) * $app->h() / 4 / 32000 ) + ( $app->h / 2 );
    my $point_x = ( $i / @stream ) * $app->w;

    # Using the parameters
    # Surface, box coordinates and color as RGBA
    SDL::GFX::Primitives::box_RGBA(
        $app,
        $point_x - $l_wdt,
        $app->h() / 2,
        $point_x + $l_wdt,

```

Chapter 9 | SOUND AND MUSIC

```
        $point_y, 40, 0, 255, 128
    );
    SDL::GFX::Primitives::box_RGBA(
        $app,
        $point_x - $l_wdt,
        $app->h() / 2,
        $point_x + $l_wdt,
        $point_y_r, 255, 0, 40, 128
    );

}

$app->flip();

};
```

Whenever a song finishes `SDL::Mixer::Music::playing_music` returns 0. We detect this change in state and call `music_finished_playing()` where the program attaches our `$play_next_song_callback` to switch to the next song gracefully:

```
my $cms_move_callback_id;
my $pns_move_callback_id;
my $play_next_song_callback;

sub music_finished_playing {
    SDL::Mixer::Music::halt_music();
    $pns_move_callback_id = $app->add_move_handler($play_next_song_callback)
        if ( defined $play_next_song_callback );
}

$play_next_song_callback = sub {
    return $app->stop() if $current_song >= @songs;
    my $song = SDL::Mixer::Music::load_MUS( $songs[ $current_song++ ] );
    SDL::Mixer::Music::play_music( $song, 0 );

    $app->remove_move_handler($pns_move_callback_id)
```

```

        if defined $pns_move_callback_id;
    };

```

A move handler is attached to detect if music is playing or not:

```

$app->add_move_handler(
    sub {
        my $music_playing = SDL::Mixer::Music::playing_music();
        music_finished_playing() unless $music_playing;
    }
)

```

The first callback to trigger the `$play_next_song_callback` gets the first song:

```

$app->add_show_handler($current_music_callback);
$pns_move_callback_id = $app->add_move_handler($play_next_song_callback);

```

... and a keyboard event handler for a keypress allows the user to move through songs:

```

$app->add_event_handler(
    sub {
        my ($event, $app) = @_;

        if( $event->type == SDL_KEYDOWN && $event->key_sym == SDLK_DOWN)
        {
            #Indicate that we are done playing the music_finished_playing
            music_finished_playing();
        }
    }
);

```

From there, the application is ready to run:

```

$app->run();

```

Chapter 9 | SOUND AND MUSIC

... and the final code gracefully stops `SDL::Mixer`:

```
SDL::Mixer::Effects::unregister( MIX_CHANNEL_POST, $music_data_effect_id );
SDL::Mixer::Music::hook_music_finished();
SDL::Mixer::Music::halt_music();
SDL::Mixer::close_audio();
```

The result? Several dozen lines of code to glue together the SDL mixer and display a real-time visualization of the music.

10

CPAN

The Comprehensive Perl Archive Network (CPAN) is the other part of the Perl language. By now most Perl developers should be aware of how to search and get modules from CPAN. This chapter will focus on why to use CPAN for games. Next we will take a look in what domain (Model, View or Controller) does a module solve a problem for. Moreover we would want to look at what is criteria to pick one module from another, using the many tools provided by CPAN.

10.1 Modules

It is good to reuse code.

10.1.1 MVC Method

See where the module fits, Model, View or Controller

View

SDL will do most but helper module (Clipboard) are cool to have.

The *SDLx::Widget* bundle comes separately, but is meant to provide you with several common game elements such as menu, dialog boxes and buttons, all seamlessly integrated with SDL.

Model

The logic and modelling behind most popular games is already on CPAN, so you can easily plug them in to create a new game of Chess, Checkers, Go, Life, Minesweeping, Cards, etc. There are even classes for platform games (like *Games::Nintendo::Mario*), creating and solving mazes, generating random dungeon maps, you name it. Have a look at *Roguelike-Utills* and *Games::RolePlay::MapGen* for just a few of those.

If your game needs to store data, like objects and status for saved games or checkpoints, you can use *Storable* or any of the many data serializers available.

In fact, speaking of data structures, it is common to keep game data in standard formats such as JSON, YAML or XML, to make you able to import/export them directly from third-party tools like visual map makers or 3D modeling software. Perl provides very nice modules to handle the most popular formats - and some pretty unusual ones. Parsers vary in speed, size and thoroughness, so make sure to check the possible candidates and use the one that fits your needs for speed, size and accuracy.

Controller

If you need to roll a dice, you can use *Games::Dice*, that even lets you receive an array of rolled dice, and use RPG-like syntax (e.g. “2d6+1” for 2 rolls of a 6-side die, adding 1 to the result).

You can also use *Sub::Frequency* if you need to do something or trigger a particular action or event only sometimes, or at a given probability.

Your game may need you to mix words, find substrings or manipulate word permutations in any way (like when playing scrabble), in which case you might find the *Games::Word* module useful.

10.2 Picking Modules

So, you thought of a nice game, identified your needs, typed some keywords in `HTTP://Search.CPAN.Org`, and got tons of results. What now? How to avoid vaporware and find the perfect solution for your needs?

10.2.1 Documentation

Once you find a potential module for your application, make sure you will know how to use it. Take a look at the SYNOPSIS section of the module, it should contain some code snippets showing you how to use the module’s main features. Are you comfortable with the usage syntax? Does it seem to do what you expect it to? Will it fit nicely to whatever it is you’re coding?

Next, skim through the rest of the documentation. Is it solid enough for you? Does it look complete enough for your needs, or is it easily extendable?

10.2.2 License

It's useless to find a module you can't legally use. Most (if not all) modules in `HTTP://Search.CPAN.Org` are free and open source software, but even so each needs a license telling developers what they can and cannot do with it. A lot of CPAN modules are released “*under the same terms as Perl itself*”, and this means you can pick between the Artistic License or the GPL (version 1).

Below is a short and incomplete list of some popular license choices by CPAN developers:

- Artistic License - `HTTP://Dev.Perl.Org/licenses/artistic.html`
- GPL (all versions and variations) - `HTTP://GNU.Org/licenses`
- MIT License - `HTTP://OpenSource.Org/licenses/mit-license.php`

See `HTTP://OpenSource.Org/licenses/alphabetical` for a comprehensive list with each license's full documentation.

You should be able to find the module's license by going to a “LICENSE AND COPYRIGHT” section, usually available at the bottom of the documentation, or by looking for a license file inside that distribution.

Note: Some modules might even be released into CPAN as *public domain*, meaning they are not covered by intellectual property rights at all, and you are free to use them as you see fit. Even so, it's usually considered polite to mention authors as a courtesy, you know, giving credit where credit is due.

10.2.3 Ratings

The CPAN Ratings is a service where developers rate modules they used for their own projects, and is a great way to have some actual feedback on how it was to use the code on a real application. The ratings are compiled into a 1 to 5 grade, and displayed below the module name on CPAN. You can click on the “Reviews” link right next to the rating stars to see any additional comments by the reviewers, praising, criticizing or giving some additional comments or the distribution and/or its competition.

10.2.4 Dependencies

Modules exist so you don’t have to reinvent the wheel, and for that same reason each usually depends on one or more modules itself. Don’t worry if a module depends on several others - code reusability is a good thing.

You may, however, be interested in **which** modules it depends on, or, more practically, in the likelihood of a clean installation by your users. For that, you can browse to `HTTP://Deps.CPANTesters.Org` and input the module’s name on the search box.

The CPAN Testers is a collaborative matrix designed to help developers test their modules in several different platforms, with over a hundred testers each month making more than 3 million reports of CPAN modules. This particular CPAN Testers service will show you a list of dependencies and test results for each of them, calculating the average chance of all tests passing (for any platform).

While seeing all the dependencies and test results of a couple of modules that do the same thing might help you make your pick, it’s important to realize that the “*chance of all tests passing*” information at the bottom of the results means very little. This is because test failures can rarely be considered independent events, and are usually tied to not running on a specific type of operating system, to the perl version, or even due to the tester running out of memory for reasons that may not even concern the module being evaluated. If you don’t care about your application running on AIX or on perl 5.6.0, why would you dismiss a module that only fails on those conditions?

10.2.5 CPAN Testers Charts

So, how do you know the actual test results for a module on the CPAN? How can you tell if that module will run in your target machine according to architecture, operating system and perl version?

The CPAN Testers website at [HTTP://CPANTesters.org](http://CPANTesters.org) offers a direct search for distributions by name or author. To see the results for the SDL module, for instance, you can go to [HTTP://CPANTesters.org/distro/S/SDL.html](http://CPANTesters.org/distro/S/SDL.html). You can also find a test report summary directly on CPAN, by selecting the distribution and looking at the “*CPAN Testers*” line. If you click on the “*View Reports*” link, you’ll be redirected to the proper CPAN Testers page, like the one shown above.

The first chart is a PASS summary, containing information about the most recent version of that module with at least one *PASS* report submitted, separated by platform and perl version.

Second is a list of selected reports, detailing all the submitted test results for the latest version of the given module. If you see a *FAIL* or *UNKNOWN* result that might concern you - usually at a platform you expect your application to run - you can click on it to see a verbose output of all the tests, to see why it failed.

Another interesting information displayed is the report summary on the left sidebar, showing a small colored graph of PASS-UNKNOWN-FAIL results for the latest versions of the chosen module. If you see a released version with lots of FAIL results, it might be interesting to dig deeper or simply require a greater version of that module in your application.

Bug Reports

When picking a module to use, it is very important to check out its bug reports. You can do that by either clicking on the “*View/Report Bugs*” link on the module’s page on CPAN, or on the “*CPAN RT*” (for Request Tracker) box on the right side of the documentation page.

Look for open bugs and their description - i.e. if it's a bug or a wishlist - and see if it concerns your planned usage for that module. Some bug reports are simple notices about a typo on the documentation or a very specific issue, so make sure you look around the ticket description to see if it's something that blocks your usage, or if you can live with it, at least until the author delivers an update.

It may also interest you to see how long the open bugs have been there. Distributions with bugs dating for more than two years might indicate that the author abandoned the module to pursue other projects, so you'll likely be on your own if you find any bumps. Of course, being free software, that doesn't mean you can't fix things yourself, and maybe even ask the author for maintainance privileges so you can update your fixes for other people to use.

10.2.6 Release Date

A old distribution might mean a solid and stable distribution, but it can also mean that the author doesn't care much about it anymore. If you find a module whose latest version is over 5 years old, make sure to double check test results and bug reports, as explained above.

10.3 Conclusion

CPAN is an amazing repository filled with nice modules ready for you to use in your games. More than often you'll find that 90% of your application is already done on CPAN, and all you have to do to get that awesome idea implemented is glue them together, worrying only about your application's own logic instead of boring sidework. This means faster development, and more fun!

10.4 Author

This chapter's content graciously provided by Breno G. de Oliveira (garu).

11

Pixel Effects

In this chapter we will look at how to use pixel effects in Perl. Pixel effects are operations that are done directly on the bank of a `SDL_Surface`'s pixel. These effects are used to do visual effects in games and applications, most notably by `Frozen Bubble`.

These effects can be done in purely in Perl, for 1 passes and non real time applications. Effects that need to be done real time will have to be done in C via XS. This chapter will show two methods of doing this.

11.1 Sol's Ripple Effect

For our first pixel effect we will be doing is a ripple effect from a well known SDL resource, [HTTP://Sol.Gfxile.Net/gp/ch02.html](http://Sol.Gfxile.Net/gp/ch02.html). This effects uses `SDL::get_ticks` to animate a ripple effect across the surface as seen in the following figure.




Figure 11.1: Sol's Chapter 01 Ripple Effect

11.1.1 Pure Perl

First lets make the effect in pure Perl. To do any operations with a `SDL::Surface` we must do `SDL::Video::lock_surface()` call as seen below. Locking the surface prevents other process in SDL from accessing the surface. The surface pixels can be accessed several ways from Perl. Here we are using the `SDL::Surface::set_pixels` which takes an offset for the `SDL::Surface` pixels array, and sets a value there for us. The actual pixel effect is just a time dependent (using `SDL::get_ticks` for time) render of a function. See [HTTP://Sol.Gfxile.Net/gp/ch02.html](http://Sol.Gfxile.Net/gp/ch02.html) for a deeper explanation.

```
1 use strict;  
2 use warnings;
```



```

3
4 use SDL;
5 use SDLx::App;
6
7     # Render callback that we use to fiddle the colors on the surface
8 sub render {
9     my $screen = shift;
10    if ( SDL::Video::MUSTLOCK($screen) ) {
11        return if ( SDL::Video::lock_surface($screen) < 0 );
12    }
13
14    my $ticks = SDL::get_ticks();
15    my ( $i, $y, $yofs, $ofs ) = ( 0, 0, 0, 0 );
16    for ( $i = 0; $i < 480; $i++ ) {
17        for ( my $j = 0, $ofs = $yofs; $j < 640; $j++, $ofs++ ) {
18            $screen->set_pixels( $ofs, ( $i * $i + $j * $j + $ticks ) );
19        }
20        $yofs += $screen->pitch / 4;
21    }
22
23
24    SDL::Video::unlock_surface($screen) if ( SDL::Video::MUSTLOCK($screen) );
25
26    SDL::Video::update_rect( $screen, 0, 0, 640, 480 );
27
28    return 0;
29 }
30
31
32 my $app = SDLx::App->new( width => 640,
33                          height => 480,
34                          eoq => 1,
35                          title => "Grovvy XS Effects" );
36
37 $app->add_show_handler( sub{ render( $app ) } );
38
39 $app->run();

```

One you run this program you will find it pretty much maxing out the CPU and not running very smoothly. At this point running a loop through the entire pixel bank of a 640x480 sized screen is too much for Perl. We will need to move the intensive calculations to c.

11.1.2 Inline Effects

In the below example we use `Inline` to write Inline c code to handle the pixel effect for us. `SDL` now provides support to work with `Inline`. The render callback is now moved to c code, using `Inline c`. When the program first runs it will compile the code and link it in for us.

```
1  use strict;
2  use warnings;
3  use Inline with => 'SDL';
4  use SDL;
5  use SDLx::App;
6
7
8  my $app = SDLx::App->new( width => 640,
9                           height => 480,
10                          eoq => 1,
11                          title => "Groovy XS Effects" );
12
13      # Make render a callback which has the expected signature from show_handlers
14  $app->add_show_handler( \&render);
15
16  $app->run();
17
18  use Inline C => <<'END';
19
20      // Show handlers recieve both float and the SDLx::App which is a SDL_Screen
21  void render( float delta, SDL_Surface *screen )
22  {
23      // Lock surface if needed
24      if (SDL_MUSTLOCK(screen))
```

```

25         if (SDL_LockSurface(screen) < 0)
26             return;
27
28         // Ask SDL for the time in milliseconds
29         int tick = SDL_GetTicks();
30
31         // Declare a couple of variables
32         int i, j, yofs, ofs;
33
34         // Draw to screen
35         yofs = 0;
36         for (i = 0; i < 480; i++)
37         {
38             for (j = 0, ofs = yofs; j < 640; j++, ofs++)
39             {
40                 ((unsigned int*)screen->pixels)[ofs] = i * i + j * j + tick;
41             }
42             yofs += screen->pitch / 4;
43         }
44
45         // Unlock if needed
46         if (SDL_MUSTLOCK(screen))
47             SDL_UnlockSurface(screen);
48
49         // Tell SDL to update the whole screen
50         SDL_UpdateRect(screen, 0, 0, 640, 480);
51     }
52
53     END

```


12

Additional Modules

12.1 PDL

The Perl Data Language (PDL) is a tool aimed at a more scientific crowd. Accuracy is paramount and speed is the name of the game. PDL brings to Perl fast matrix and numerical calculations. For games in most cases a accuracy is not critical, but speed and efficiency is a great concern. For this reason we will briefly explore how to share SDL texture data between PDL and OpenGL.

This example will do the following:



Figure 12.1: Not terribly interesting, but the speed is phenomenal

12.1.1 Make the application

Let's start an application to use with PDL. Make sure you do use PDL.

```
+ use strict;
+ use warnings;
+ use SDL;
+ use SDL::Video;
+ use SDLx::App;
+
+ use PDL;
+
+ my $app = SDLx::App->new(
+     title => 'PDL and SDL application',
+     width => 640, height => 480, depth => 32,
+     eoq => 1);
```

12.1.2 Attaching the Piddle

PDL core object is something called a piddle. To be able to perform PDL calculations and show them on SDL surfaces, we need to share the memory between them. SDL Surface memory is stored in a `void *` block called `pixels`. `void *` memory has the property that allows Surfaces to have varying depth, and pixel formats. This also means that we can have PDL's memory as our `pixels` for our surface.

```
+ sub make_surface_piddle {  
+ my ( $bytes_per_pixel, $width, $height ) = @_;  
+ my $piddle = zeros( byte, $bytes_per_pixel, $width, $height );  
+ my $pointer = $piddle->get_dataref();
```

At this point we have a pointer to the `$piddle`'s memory with the given specifications. Next we have our surface use that memory.

```
+ my $s = SDL::Surface->new_form(  
+                                     $pointer, $width, $height, 32,  
+                                     $width * $bytes_per_pixel  
+                                     );  
+  
+ #Wrap it into a SDLx::Surface for ease of use  
+ my $surface = SDLx::Surface->new( surface => $s );  
+  
+ return ( $piddle, $surface );  
+ }
```

Lets make some global variables to hold our `$piddle` and `$surface`.

```
+ my ( $piddle, $surface ) = make_surface_piddle( 4, 400, 200 );
```

12.1.3 Drawing and Updating

`make_surface.piddle()` will return to use an anonymous array with a `$piddle` and `$surface` which we can use with PDL and SDL. PDL will be used to operate on the `$piddle`. SDL will be used to update the `$surface` and render it to the `SDLx::App`.

```
+ $app->add_move_handler( sub {  
+  
+   SDL::Video::lock_surface($surface);  
+  
+   $piddle->mslice( 'X',  
+     [ rand(400), rand(400), 1 ],  
+     [ rand(200), rand(200), 1 ]  
+   ) .= pdl( rand(225), rand(225), rand(225), 255 );  
+  
+   SDL::Video::unlock_surface($surface);  
+ } );
```

`SDL::Video::lock_surface` prevents SDL from doing any operations on the `$surface` until `SDL::Video::unlock_surface` is called. Next we will blit this surface onto the `$app`.

In this case we use PDL to draw random rectangles of random color.

12.1.4 Running the App

Finally we blit the `$surface` and update the `$app`.

```
+ $app->add_show_handler( sub {  
+  
+   $surface->blit( $app, [0,0,$surface->w,$surface->h], [10,10,0,0] );  
+   $app->update();  
+  
+ } );
```



```
+ $app->run();
```

12.1.5 Complete Program

```
1  use strict;
2  use warnings;
3  use SDLx::App;
4
5  use PDL;
6
7  my $app = SDLx::App->new(
8      title => "PDL and SDL application",
9      width => 640, height => 480, eoq => 1 );
10
11
12  sub make_surface_piddle {
13      my ( $bytes_per_pixel, $width, $height ) = @_;
14      my $piddle = zeros( byte, $bytes_per_pixel, $width, $height );
15      my $pointer = $piddle->get_dataref();
16      my $s = SDL::Surface->new_from(
17          $pointer, $width, $height, 32,
18          $width * $bytes_per_pixel
19      );
20
21      my $surface = SDLx::Surface->new( surface => $s );
22
23      return ( $piddle, $surface );
24  }
25
26
27  my ( $piddle, $surface ) = make_surface_piddle( 4, 400, 200 );
28
29  $app->add_move_handler( sub {
30
31      SDL::Video::lock_surface($surface);
```

```

32
33     $piddle->mslice( 'X',
34         [ rand(400), rand(400), 1 ],
35         [ rand(200), rand(200), 1 ]
36     ) .= pdl( rand(225), rand(225), rand(225), 255 );
37
38     SDL::Video::unlock_surface($surface);
39 } );
40
41
42 $app->add_show_handler( sub {
43
44     $surface->blit( $app, [0,0,$surface->w,$surface->h], [10,10,0,0] );
45     $app->update();
46
47 });
48
49 $app->run();

```

12.2 OpenGL and SDL

OpenGL is a cross platform library for interactive 2D and 3D graphics applications. However OpenGL specifies only the graphics pipeline and doesn't handle inputs and events. SDL can hand over the graphics component of an application over to OpenGL and take control over the event handling, sound, and textures. In the first example we will see how to set up Perl's `OpenGL` module with `SDL::App`.




Figure 12.2: The lovely blue teapot

12.2.1 SDL Setup

```
use strict;
use warnings;
use SDL;
use SDLx::App;

use OpenGL qw/:all/;

my $app = SDLx::App->new(
    title => "OpenGL App",
    width => 600,
    height => 600,
    gl    => 1,
    eoq   => 1
);
```

Chapter 12 | ADDITIONAL MODULES

```
$app->run();
```

Enabling OpenGL mode is as simple as adding the `gl` flag to the `SDLx::App` constructor.

12.2.2 OpenGL Setup

Next we will make a OpenGL perspective with the `$app`'s dimensions:

```
glEnable(GL_DEPTH_TEST);  
glMatrixMode(GL_PROJECTION);  
glLoadIdentity();  
gluPerspective(60, $app->w/$app->h, 1, 1000 );  
glTranslatef( 0,0,-20);
```

Additionally we will be initializing `glut`, but just to draw something quick.

```
#Using glut to draw something interesting really quick  
glutInit();
```

12.2.3 The Render Callback

Now we are prepared to put something on the screen.

```
$app->add_show_handler(  
    sub{  
        my $dt = shift;  
  
        #clear the screen  
        glClear( GL_COLOR_BUFFER_BIT | GL_DEPTH_BUFFER_BIT );  
        glColor3d(0,1,1);
```

```

        glutSolidTeapot(2);

        #sync the SDL application with the OpenGL buffer data
        $app->sync;

    }

);

```

At this point there should be a light blue teapot on the screen. The only special thing to notice here is that we need to call the `sync()` method on `$app`. This will flush the buffers and update the SDL application for us.

12.2.4 Event handling

Event handling is the same as any other `SDLx::App`. We will use the mouse motion changes to rotate the teapot.

First add a global variable to hold your rotate values. And then use those values to rotate our teapot.

```

glutInit();

+ my $rotate = [0,0];

$app->add_show_handler(
    sub{
        my $dt = shift;

        #clear the screen
        glClear( GL_COLOR_BUFFER_BIT | GL_DEPTH_BUFFER_BIT );
        glColor3d(0,1,1);

+   glPushMatrix();

```

Chapter 12 | ADDITIONAL MODULES

```
+         glRotatef($rotate->[0], 1,0,0);
+ glRotatef($rotate->[1], 0,1,0);

glutSolidTeapot(2);

#sync the SDL application with the OpenGL buffer data
$app->sync;

glPopMatrix();
}
);
```

Next we will add an event handler to the app to update the rotate values for us.

```
$app->add_event_handler(

    sub {
        my ($e ) = shift;

        if( $e->type == SDL_MOUSEMOTION )
        {
            $rotate =  [$e->motion_x, $e->motion_y];
        }

    }

);
```

Finally we run the application.

```
$app->run();
```

12.2.5 Complete Code

```
1  use strict;
2  use warnings;
3  use SDL;
4  use SDLx::App;
5  use SDL::Event;
6
7  use OpenGL qw/:all/;
8
9  my $app = SDLx::App->new(
10      title => "OpenGL App",
11      width => 600,
12      height => 600,
13      gl => 1,
14      eoq => 1
15  );
16
17  glEnable(GL_DEPTH_TEST);
18  glMatrixMode(GL_PROJECTION);
19  glLoadIdentity;
20  gluPerspective(60, $app->w/$app->h, 1, 1000 );
21  glTranslatef( 0,0,-20);
22  glutInit();
23
24  my $rotate = [0,0];
25
26  $app->add_show_handler(
27      sub{
28          my $dt = shift;
29
30          #clear the screen
31          glClear( GL_COLOR_BUFFER_BIT | GL_DEPTH_BUFFER_BIT );
32          glColor3d(0,1,1);
33
34          glPushMatrix();
35
```

Chapter 12 | ADDITIONAL MODULES

```
36         glRotatef($rotate->[0], 1,0,0);
37         glRotatef($rotate->[1], 0,1,0);
38
39         glutSolidTeapot(2);
40
41     #sync the SDL application with the OpenGL buffer data
42     $app->sync;
43
44     glPopMatrix();
45     }
46     );
47
48     $app->add_event_handler(
49
50         sub {
51             my ($e ) = shift;
52
53             if( $e->type == SDL_MOUSEMOTION )
54             {
55                 $rotate =  [$e->motion_x,  $e->motion_y];
56             }
57
58         }
59
60     );
61
62     $app->run();
```


13

Free Resources

When developing a game, coding is unfortunately not everything. Not by a very, very long shot. To make up (a little) for that, below is a list of free resources you can use in your games, either in full or simply as inspiration for your own productions, in case you have an artistic vein yourself.

Make sure to check the licence for the resource and use it accordingly, giving the original author proper credit.

Note: websites come and go, so if you find any of the links broken, or know a nice free resource that's not listed here, please let us know so we can update the list.

13.1 Art and Sprites

- [HTTP://CGTextures.Com](http://CGTextures.Com)
- [HTTP://Mayang.Com/textures](http://Mayang.Com/textures)
- [HTTP://GRSites.Com/archive/textures](http://GRSites.Com/archive/textures)
- [HTTP://ImageAfter.Com](http://ImageAfter.Com)
- [HTTP://AbsoluteCross.Com/graphics/textures](http://AbsoluteCross.Com/graphics/textures)
- [HTTP://FreeFoto.Com](http://FreeFoto.Com)
- [HTTP://Noctua-Graphics.De](http://Noctua-Graphics.De)
- [HTTP://M3Corp.Com/a/download/3d_textures/pages](http://M3Corp.Com/a/download/3d_textures/pages)
- [HTTP://ReinersTileSet.4Players.De/englisch.html](http://ReinersTileSet.4Players.De/englisch.html)
- [HTTP://VirtualWorlds.Wikia.Com](http://VirtualWorlds.Wikia.Com)
- [HTTP://Lunar.LostGarden.Com/labels/free%20game%20graphics.html](http://Lunar.LostGarden.Com/labels/free%20game%20graphics.html)
- [HTTP://PDGameResources.WordPress.Com](http://PDGameResources.WordPress.Com)
- [HTTP://GamingGroundZero.Com](http://GamingGroundZero.Com)
- [HTTP://FlyingYogi.Com/fun/spritelib.html](http://FlyingYogi.Com/fun/spritelib.html)
- [HTTP://PixelPoke.Com](http://PixelPoke.Com)

13.2 Music and Sound Effects

- [HTTP://FreeSound.Org](http://FreeSound.Org)
- [HTTP://CCMixer.Org](http://CCMixer.Org)
- [HTTP://Jamendo.Com](http://Jamendo.Com)
- [HTTP://8BC.Org](http://8BC.Org)
- [HTTP://Sakari-Infinity.Net](http://Sakari-Infinity.Net)
- [HTTP://FindSounds.Com](http://FindSounds.Com)
- [HTTP://GRSites.Com/archive/sounds](http://GRSites.Com/archive/sounds)

13.3 Fonts

- [HTTP://DAFont.Com](http://DAFont.Com)
- [HTTP://FontSquirrel.Com](http://FontSquirrel.Com)
- [HTTP://TheLeagueOfMoveableType.Com](http://TheLeagueOfMoveableType.Com)
- [HTTP://OpenFontLibrary.Org](http://OpenFontLibrary.Org)
- [HTTP://AcidFonts.Com](http://AcidFonts.Com)
- [HTTP://GRSites.Com/archive/fonts](http://GRSites.Com/archive/fonts)
- [HTTP://UrbanFonts.Com](http://UrbanFonts.Com)

13.4 DIY

[HTTP://GameSoundDesign.Com](http://GameSoundDesign.Com) has several tips on making game music, including several sources for inspiration.

If you want to create 3D models, either for cutscenes or to integrate into your game via OpenGL, there are several nice libraries out there for you:

Blender - A free 3D graphics application for modeling, texturing, water and smoke simulations, rendering, etc. [HTTP://Blender.Org](http://Blender.Org)

OGRE - An open-source graphics rendering engine, used in a large number of production projects. It can be easily integrated via Scott Lanning's *Ogre* Perl bindings, on CPAN. [HTTP://Ogre3D.Org](http://Ogre3D.Org)

13.5 Author

This chapter's content graciously provided by Breno G. de Oliveira (garu).

Index

List::Util, 74
SDL::Event, 27
SDL_ACTIVEEVENT, 28
SDL_JOYAXISMOTION, 28
SDL_JOYBALLMOTION, 28
SDL_JOYBUTTONDOWN, 28
SDL_JOYBUTTONUP, 28
SDL_JOYHATMOTION, 28
SDL_KEYDOWN, 28
SDL_KEYUP, 28
SDL_MOUSEBUTTONDOWN, 28
SDL_MOUSEBUTTONUP, 28
SDL_MOUSEMOTION, 28
SDL_QUIT, 28
SDL_SYSWMEVENT, 28
SDL_USEREVENT, 28
SDL_VIDEOEXPOSE, 28
SDL_VIDEORESIZE, 28
SDLx::Controller, 28
SDLx::Text, 64

Allan Alcorn, 49
Atari, 49

circle, 20

color, 15

displacement, 53

event, 27
events; queue, 27

FPS, 40
frame, 40

game loop, 37

Inline, 130

movement, 53

Pong, 49
primitives, 17

rectangle, 19

Screen, 9
Separating Axis Theorem, 60
Surface, 9

velocity, 53
Video, 9
Video Device, 9